بِسْمِ اللَّهِ الرَّحْمَٰنِ الرَّحِيمِ

KOMPUTASI SIMBOLIK DAN NUMERIK MENGGUNAKAN SAGEMATH

Oleh:
Subiono
Departemen Matematika, Institut Teknologi Sepuluh Nopember
subiono2008@matematika.its.ac.id

17 Oktober 2020

TENTANG SAGEMATH

1. **Apa SageMath itu?**
2. **Sejarah dan Pengembangan SageMath**
3. **Server SageMath di Departemen Matematika-ITS**
4. **Contoh-contoh masalah Matematika diselesaikan dengan SageMath:**

   - [**Tripel Pythagoras** (a, b, c) (http://localhost:8888/notebooks/Tripel_Pythagoras_2020.ipynb)](http://localhost:8888/notebooks/Tripel_Pythagoras_2020.ipynb)
   - [**Masalah Analisis Matematika** (http://localhost:8888/notebooks/Masalah_Analisis_Matematika_2020.ipynb)](http://localhost:8888/notebooks/Masalah_Analisis_Matematika_2020.ipynb)
   - [**Kalkulus** (http://localhost:8888/notebooks/Kalkulus_2020.ipynb)](http://localhost:8888/notebooks/Kalkulus_2020.ipynb)
   - [**Integral** (http://localhost:8888/notebooks/Integral_2020.ipynb)](http://localhost:8888/notebooks/Integral_2020.ipynb)
   - [**Metode Newton Raphson** (http://localhost:8888/notebooks/Metode_NewtonRaphson_2020.ipynb)](http://localhost:8888/notebooks/Metode_NewtonRaphson_2020.ipynb)
   - [**Contoh Animasi Sikloida** (http://localhost:8888/notebooks/Animasi_Sikloida_2020.ipynb)](http://localhost:8888/notebooks/Animasi_Sikloida_2020.ipynb)
   - [**Bilangan *Floating Point*** (http://localhost:8888/notebooks/Bilangan_Floating_Point_2020.ipynb)](http://localhost:8888/notebooks/Bilangan_Floating_Point_2020.ipynb)

In [ ]:

```

```

PENGENALAN SageMath

Nama awal SageMath adalah **SAGE** yaitu "***System for Algebra and Geometry Experimentation***". Sekarang berubah menjadi **SageMath**. SageMath adalah suatu perangkat lunak gratis (*open source*) sebagai alternatif open source **MAPLE**, **MATHEMATICA** dan **MATLAB**. Banyak paket program open source matematika berkualitas tinggi tercakup dalam **SageMath** yang merupakan berbagai libraries dari proyek *open source* lain. Yaitu untuk

1. **Aljabar** : [GAP (https://www.gap-system.org/)](https://www.gap-system.org/) , [Singular (https://www.singular.uni-kl.de/)](https://www.singular.uni-kl.de/), [FLINT (http://flintlib.org/)](http://flintlib.org/).

2. **Geometri Aljabar** : Singular (https://www.singular.uni-kl.de/).
3. **Aritmetika Sebarang Presisi** : MPIR (http://mpir.org/), MPFR (https://www.mpfr.org/), MPFI (https://gforge.inria.fr/frs/?group_id=157), NTL (https://shoup.net/ntl/), mpmath (http://mpmath.org/), Arb (http://arblib.org/).
4. **Geometri Arimetika** : PARI/GP (https://pari.math.u-bordeaux.fr/), NTL (https://shoup.net/ntl/), mwrank (http://homepages.warwick.ac.uk/~masgaj/mwrank/), ECM (https://en.wikipedia.org/wiki/Lenstra_elliptic-curve_factorization#GMP-ECM_and_EECM-MPFQ).
5. **Kalkulus** : Maxima (http://maxima.sourceforge.net/), SymPy (https://github.com/sympy/), GiNaC (https://www.ginac.de/), Giac (https://sourceforge.net/projects/xcas/), FriCAS (fricas.sourceforge.net/).
6. **Kombinatorik** : Symmetrica (), Sage-Combinat (https://wiki.sagemath.org/combinat).
7. **Aljabar Linear** : ATLAS (http://math-atlas.sourceforge.net/), BLAS (https://www.netlib.org/blas/), LAPACK (http://performance.netlib.org/lapack/), NumPy (https://numpy.org/), LinBox (https://github.com/linbox-team/linbox), IML (https://orms.mfo.de/project?terms=&id=311), GSL (https://www.gnu.org/software/gsl/doc/html/linalg.html).
8. **Teori Graf** : NetworkX (http://networkx.github.io/).
9. **Teori Grup** : GAP (https://www.gap-system.org/).
10. **Komputasi Numerik** : GSL (https://www.gnu.org/software/gsl/doc/html/linalg.html), SciPy (https://www.scipy.org/), NumPy (https://numpy.org/), ATLAS (http://math-atlas.sourceforge.net/).
11. **Teori Bilangan** : PARI/GP (https://pari.math.u-bordeaux.fr/), FLINT (http://flintlib.org/), NTL (https://shoup.net/ntl/).
12. **Komputasi Statistika** : R (https://www.r-project.org/), SciPy (https://www.scipy.org/).
13. **Command-line shell** : IPython (http://ipython.org/).
14. **Database** : ZODB (http://www.zodb.org/en/latest/), SQLite (https://sqlite.org/download.html).
15. **Graphical interface** : SageMath Notebook (https://wiki.sagemath.org/notebook), jsMath (http://www.math.union.edu/~dpvc/jsmath/welcome.html).
16. **Graphics** : matplotlib (https://matplotlib.org/), Tachyon (jedi.ks.uiuc.edu/~johns/tachyon/), GD (https://libgd.github.io/), Jmol (http://jmol.sourceforge.net/).
17. **Interactive Programming Language** : Python (https://www.python.org/).
18. **Networking** : Twisted (https://github.com/twisted/twisted).
19. **Differential Geometry dan Tensor Calculus** : Sage Manifolds (https://github.com/sagemanifolds/SageManifolds).

SageMath menggunakan suatu syntax mirip **Phyton** yang mendukung konstruksi procedural, functional dan object-oriented. Beberapa keutamaan SageMath adalah :

- Notebook berbasis suatu browser untuk tinjauan dan penggunaan kembali masukan dan keluaran sebelumnya, mencakup grafik dan anotasi teks. Kompatibel dengan Firefox, Opera, Konqueror, Google Chrome and Safari. Notebook bisa diakses secara lokal atau jarak jauh dan koneksinya dapat diamankan dengan HTTPS.
- Command-line interface berbasis suatu teks menggunakan IPython.
- Mendukung parallel processing menggunakan multi-core processors, multiple processors, atau distributed computing.
- Kalkulus menggunakan Maxima dan SymPy.
- Aljabar Linear Numerik menggunakan GSL, SciPy dan NumPy.
- Mencakup berbagai Libraries fungsi-fungsi matematika elementary dan special.

- Fungsi simbolik grafik 2D dan 3D serta data numerik.
- Manipulasi Matrix, meliputi sparse arrays.
- Libraries Multivariate statistics, menggunakan R dan SciPy.
- Toolkit untuk menambahkan user interfaces ke kalkulasi dan aplikasi.
- Visualisasi dan alat analisis Graph theory.
- Berbagai fungsi Libraries dari teori bilangan.
- Mendukung complex numbers, arbitrary precision dan symbolic computation.
- Pengola kata teknis mencakup formula editing dan embedding SageMath dalam dokumen LaTeX.
- Library baku Python, mencakup alat untuk menghubungkan SQL, HTTP, HTTPS, NNTP, IMAP, SSH, IRC, FTP dan yang lainnya.
- Antarmuka ke beberapa aplikasi pihak-ketiga (third-party) seperti Mathematica, Magma, R, dan Maple.
- Dokumentasi menggunakan Sphinx.
- Suatu automated test-suite.
- Kode eksekusi Fortran, C, C++, dan Cython.
- SageMath dapat menarik (*can pull up*) Mathematica dalam suatu program. Interfacing dengan cara ini didokumentasi secara resmi ke Sage.

Sejarah dan Pengembangan SageMath

Versi pertama **SageMath** dirilis pada 24 Februari 2005 sebagai perangkat lunak babas dan open-source berdasarkan lisensi dari GNU, *General Public License versi* 3, dengan tujuan awal untuk menciptakan sebuah "alternatif open source untuk **Magma**, **Maple**, **Mathematica**, dan **MATLAB**". **Magma** adalah sistem aljabar komputer yang dirancang untuk memecahkan masalah dalam aljabar, teori bilangan, geometri dan kombinatorik. Perangkat lunak ini dinamai magma struktur aljabar dan berjalan pada sistem operasi mirip Unix, dan juga Windows. Magma walaupun bukan perangkat lunak non-komersial (*Cost recovery (non-commercial proprietary*)), tetapi bersifat tertutup [Magma (https://en.wikipedia.org/wiki/Magma_(computer_algebra_system))](https://en.wikipedia.org/wiki/Magma_(computer_algebra_system)). Pencetus dan pemimpin proyek **SageMath**, **William Stein**, adalah seorang matematikawan dari University of Washington.

William Stein menyadari ketika merancang **SageMath** bahwa ada banyak paket perangkat lunak matematika open source yang sudah ditulis dalam bahasa yang berbeda, yaitu C, C ++, Common Lisp, Fortran dan Python.

Daripada menemukan kembali kemudi, **SageMath** (yang sebagian besar ditulis dengan Python dan Cython) mengintegrasikan banyak paket perangkat lunak matematika khusus ke dalam antarmuka umum, yang pengguna hanya perlu mengetahui Python. Namun, **SageMath** berisi ratusan ribu baris kode unik yang menambahkan fungsi baru dan menciptakan antarmuka di antara komponennya.

SageMath menggunakan kedua siswa dan profesional untuk pengembangan. Perkembangan SageMath didukung oleh kerja sukarela dan hibah. Namun, baru pada tahun 2016 pengembang SageMath full-time pertama dipekerjakan (didanai oleh hibah UE). Pada tahun yang sama, Stein menggambarkan kekecewaannya dengan kurangnya dana akademis dan kredensial untuk pengembangan perangkat lunak, dengan alasan sebagai alasan keputusannya untuk meninggalkan posisi akademisnya untuk bekerja penuh waktu dalam proyek di sebuah perusahaan yang baru didirikan, SageMath, Inc. Untuk informasi lengkap dan sebagai referensi tentang **Sagemath** dapat diakse di [WikiSageMath (https://en.wikipedia.org/wiki/SageMath)](https://en.wikipedia.org/wiki/SageMath) dan website resmi di [https://www.sagemath.org/ (https://www.sagemath.org/)](https://www.sagemath.org/)

SageMath dapat digunakan untuk pengkajian kalkulus, teori bilangan mulai dari yang elementer hingga lanjut, kriptograpi, aljabar komutatif, teori grup, teori grap dan aljabar linear baik yang eksak maupun numerik dan berbagi hal bidang matematika yang lainnya. Selain itu SAGEMath mempunyai kemampuan grafis, animasi grafis dan program interaktif yang mudah dipahami sehingga mudah membuatnya. Dengan demikian akan sangat membantu dalam proses pembelajaran, khususnya matematika. Versi terbaru saat ini adalah 9.1 pengembangan untuk versi yang berikutnya dapat dikuti di [Perkembangan Versi SageMath (https://git.sagemath.org/sage.git/)](https://git.sagemath.org/sage.git/).

SERVER SAGEMATH DEPARTEMEN MATEMATIKA-ITS

SageMath yang ada diserver Departemen Matematika dapat diakses dengan IP address:

[SageMat Departemen Matematika-ITS (http://10.112.1.4:8080/)](http://10.112.1.4:8080/)

Pilih alamat diatas, akan muncul tampilan gambar berikut:



Gambar 1 : Tampilan IP address SageMath Departemen Matematika I
TS

Untuk membuat account, double klik **Sign up for a new Sage Notebook account** akan muncul gambar berikut



Gambar 2: Tampilan membuat suatu account

Isikan ***username*** sesuai yang diingikan, juga isikan ***password*** sesuai aturan yaitu minimal 4 karakter selanjutnya ketik ulang ***password*** yang telah disikan dan klik ***Create account***, muncul tampilan sebagaimana diberikan oleh Gambar 1. Selanjutnya ketik isian Username, Password yang telah dibuat dan klik **Sign in** sehingga muncul tampilan gambar SAGE Notebook sebagaima diberikan oleh gambar berikut.



Gambar 3 : SageMath Notebook

Untuk memulai lembar kerja baru dalam SAGE, klik ***New Worksheet*** sehingga muncul tampilan Sage Shell sebagaimana gambar berikut.



Gambar 4: Mengganti nama Worksheet

Ganti Untitled, misalkan dengan **coba** selanjutnya klik ***Rename*** akan muncul gambar tampilan Sage Notebook dan Sage Shell sebagai berikut.



Gambar 5: SageMath Notebook dan Sage Shell

Sage ***Shell*** adalah tempat untuk melakukan komputasi dengan mengetikan perintah-perintah yang dikenal oleh Sage. Pada bagian berikutnya diberikan Perintah-perintah dasar dalam SageMath yang dilakukan dalam **SageMath *Shell***.

[Kembali ke Menu]

```
In [ ]:
```

Tripel Pythagoras (a, b, c)

Masalah **tripel Pythagoras** adalah suatu masalah dalam teori bilangan. Misalkan (a, b, c) adalah tiga penyelesaian bilangan bulat positip untuk Persamaan Pythagoras. Maka bilangan-bilangan tersebut memenuhi persamaan:

$$a^2 + b^2 = c^2. \qquad (1)$$

Jika a, b, c memiliki faktor persekutuan $\lambda \neq 0$, maka

$$\left(\frac{a}{\lambda}\right)^2 + \left(\frac{b}{\lambda}\right)^2 = \left(\frac{c}{\lambda}\right)^2.$$

Jadi tanpa kehilangan sifat umum kita dapat mengasumsikan bahwa a, b, c tidak memiliki faktor persekutuan. Tripel yang demikian ini yang kita sebut primitif dalam Teorema Pythagoras. Tripel Pythagoras yang kita bahas ini semuanya primitif. Pencarian komputer cepat menghasilkan daftar berikut dari beberapa tripel Pythagoras:

$(3, 4, 5), (5, 12, 13), (8, 15, 17), (7, 24, 25), (20, 21, 29), (12, 35, 37), (9, 40, 41), (28, 45, 53), \cdots$

Tujuan kami di bagian ini adalah untuk menemukan semua solusi Persamaan primitif (1). Karena $\mathrm{fpb}(a, b, c) = 1$, jelas bahwa tidak semua a, b, c adalah genap. Sedangkan beberapa kemungkinan yang lainnya adalah:

- a, b dan c semuanya ganjil. Hal ini tidak mungkin sebab $a^2 + b^2$ adalah genap sedangkan c^2 adalah ganjil.
- a, b ganjil dan c genap. Jika a, b ganjil, maka $a^2 \equiv 1 \mod 4$, dan $b^2 \equiv 1 \mod 4$; maka $a^2 + b^2 \equiv 2 \mod 4$. Tapi karena c adalah genap, maka $4|c^2$, jadi $c^2 \equiv 0 \mod 4$. Jadi hal ini tidak mungkin juga sebab disatu sisi $c^2 = a^2 + b^2 \equiv 2 \mod 4$.
- a genap, b ganjil dan c ganjil.
- a ganjil, b genap dan c ganjil.

Kita akan melihat sejenak bahwa dua kasus terakhir ini sebenarnya mungkin. Dengan kesimetrian kita dapat mengasumsikan bahwa b adalah genap, dan a ganjil. Sehingga didapat

$$b^2 = c^2 - a^2 = (c + a)(c - a).$$

Disini, baik a dan c adalah ganjil, jadi $c + a$ dan $c - a$ adalah genap dan perkaliannya dibagi dengan 4, didapat

$$\left(\frac{b}{2}\right)^2 = \left(\frac{c+a}{2}\right)\left(\frac{c-a}{2}\right).$$

Misalkan p adalah pembagi persekutuan dari $\frac{c+a}{2}$ dan $\frac{c-a}{2}$, maka $p|\frac{c+a}{2} + \frac{c-a}{2} = c$ dan $p|\frac{c+a}{2} - \frac{c-a}{2} = a$. Jadi p adalah pembagi persekutuan dari a dan b, tetapi karena a dan b adalah koprima yaitu $\mathrm{fpb}(a, b) = 1$, maka didapat $p = 1$. Jadi $\mathrm{fpb}\left(\frac{c+a}{2}, \frac{c-a}{2}\right) = 1$, akibatnya juga $\mathrm{fpb}(c + a, c - a) = 1$. Karena hasil kali bilangan koprima $c + a$ dan $c - a$ adalah kuadrat dari b^2, maka masing-masing secara individual adalah kuadrat, yaitu, ada bilangan bulat koprima x, y sedemikian rupa sehingga

$$c + a = x^2, c - a = y^2.$$

Dengan menyelesaikan persamaan untuk a dan c didapat

$$\begin{cases} a = \dfrac{x^2 - y^2}{2} \\ b = xy \\ c = \dfrac{x^2 + y^2}{2}. \end{cases}$$

Dalam hal ini bisa ditunjukkan bahwa $a^2 + b^2 = c^2$ sebagai berikut:

$$\begin{aligned} a^2 + b^2 &= \left(\frac{x^2 - y^2}{2}\right)^2 + (xy)^2 \\ &= \left(\frac{x^2}{2}\right)^2 + \left(\frac{y^2}{2}\right)^2 - 2\left(\frac{x^2}{2}\right)\left(\frac{y^2}{2}\right) + 4\left(\frac{x^2}{2}\right)\left(\frac{y^2}{2}\right) \\ &= \left(\frac{x^2}{2}\right)^2 + \left(\frac{y^2}{2}\right)^2 + 2\left(\frac{x^2}{2}\right)\left(\frac{y^2}{2}\right) \\ &= \left(\frac{x^2 + y^2}{2}\right)^2 \\ &= c^2. \end{aligned}$$

Untuk menghindari kesan bentuk pecahan kita kalikan dengan 2 tripel Pythagoras sehingga didapat

$$\begin{cases} a = x^2 - y^2 \\ b = 2xy \\ c = x^2 + y^2. \end{cases}$$

Sebagai contoh, jika $(x, y) = (2, 1)$ kita dapatkan tripel Pythagoras terkenal $(3, 4, 5)$, dan jika $(x, y) = (3, 2)$, maka kita mendapatkan $(5, 12, 13)$. Penyelesaian umum Pythagoras kita tulis sebagai

$$(x^2 - y^2, 2xy, x^2 + y^2).$$

Pembahasan tripel Pythagoras ini kita implentasikan secara interaktif menggunakan GeoGebra dan bisa diakases di: https://www.geogebra.org/m/m43sfcfr (https://www.geogebra.org/m/m43sfcfr). Selanjutnya pembahasan tripel Pythagoras ini kita implentasikan secara interaktif dalam program **SageMath** sebagaimana berikut ini:

In [2]:

```
@ interact
def _(n=input_box(3, width=9, label="Input $n\\geq 3$:")):
    Pasangan=[(x,y) for x in range(1,n) for y in range (x+1,n)\
              if (gcd(x,y)==1)\
              and not (mod(x,2) == mod(y,2))]
    for pairs in Pasangan:
        a = pairs[1]^2 - pairs[0]^2; b = 2*pairs[0]*pairs[1];\
        c = pairs[0]^2 + pairs[1]^2
    pretty_print(html("Apakah $\\pmb{%s^2+%s^2}$\
    sama dengan $\\pmb{%s^2}$?\
    ${\\color{blue}{\\mathrm{%s}}}$"%(a,b,c,a^2+b^2==c^2)))
    pretty_print(html('Tripel primitif Pythagoras\
    $\\pmb{%s}$ adalah suatu suatu\
    segi tiga siku-siku dengan luas: %s'%((a,b,c),a*b/2)))
```

Input $n \geq 3$: 4

Apakah $\pmb{5^2+12^2}$ sama dengan $\pmb{13^2}$? ${\color{blue}{\mathrm{True}}}$

Tripel primitif Pythagoras $\pmb{(5, 12, 13)}$ adalah suatu suatu segi tiga siku-siku dengan luas: 30

[Kembali ke File Utama (http://localhost:8888/notebooks/Workshop_Utama_Unsyiah_2020.ipynb)](http://localhost:8888/notebooks/Workshop_Utama_Unsyiah_2020.ipynb)

In [ ]:

Masalah Analisis Matematika

1. Menyelesaikan pertidaksamaan:

1. $3x + 6 < -21$
2. $x^2 + 2x - 35 < 0$

In [1]:

```
# 1. Menyelesaikan pertidaksamaan 3x+6<-21

var('x')
sl=solve(3*x+6<-21,x)
pretty_print(html("Penyelesaian dari pertidaksamaan $3x+6<-21$ adalah :\
$%s$"%latex(sl)))
p=plot(3*x+27,(x,-15,10),legend_label="$f(x)=3x+27$",thickness=3)
p.show(aspect_ratio=1/4,figsize=4)
```

Penyelesaian dari pertidaksamaan $3x + 6 < -21$ adalah :$[[x < (-9)]]$



In [2]:
```
# 2. Menyelesaikan pertaksamaan x^2+2*x-35 < 0

var('x')
p=plot(x^2+2*x-35,(x,-12,10),ymax=30,legend_label="$f(x)=x^2+2x-35$",\
thickness=3)
p2=plot(x^2+2*x-35,(x,-7,5),ymax=30,fill = lambda x: 0,\
fillcolor = 'red')
p1=point([(-7,0),(5,0)],pointsize=50,rgbcolor='red')
p3=text("Daerah Penyelesaian",(-1,-6))
p4=text("-  -  -  -  -  -  -  -  -  - ",(-1,2))
p5=text("+ + + +",(8,2))
p6=text("+ + + +",(-10,2))
(p+p1+p2+p3+p4+p5+p6).show(aspect_ratio=1/4,figsize=4,gridlines='True')
sl=solve(x^2+2*x-35<0,x)
pretty_print(html("Penyelesaian $x^2+2x-35<0$ adalah :\
$%s$"%latex(sl)))
```



Penyelesaian $x^2 + 2x - 35 < 0$ adalah :$[[x > (-7), x < 5]]$

2. Nilai Mutlak

Nilai mutlak dari $x \in \mathbb{R}$ ditulis $|x|$ didefisikan oleh

$$|x| \stackrel{\text{def}}{=} \begin{cases} x, & \text{bila } x \geq 0 \\ -x, & \text{bila } x < 0\,. \end{cases}$$

Contoh

$|2| = 2, |0| = 0$ dan $|-5| = 5$.

Catatan bahwa $|x| \geq 0$ untuk setiap $x \in \mathbb{R}$ dan $|x| = 0$ bila dan hanya bila $x = 0$. Nilai $\sqrt{x} \geq 0$ untuk $x \geq 0$, jadi $|x| = \sqrt{x^2}$ untuk setiap $x \in \mathbb{R}$. Secara geometri, $|x|$ menyatakan panjang atau jarak dari x ke titik asal 0 pada garis bilangan real. Jadi

$$|a-b| = \begin{cases} a-b, & \text{bila } a \geq b \\ b-a, & \text{bila } a < b\,. \end{cases}$$

Garis bilangan real dari $|a - b|$ diberikan oleh gambar berikut:

SIFAT NILAI MUTLAK dan AKAR KUADRAT

Untuk setiap $a, b \in \mathbb{R}$ berlaku :

1. $|-a| = |a|$
2. $|ab| = |a|\,|b|$ dan $|\frac{a}{b}| = \frac{|a|}{|b|}$ dengan $b \neq 0$.
3. $|a \pm b| \leq |a| + |b|$.
4. $\sqrt{a^2} = |a|$
5. Bila $a \geq 0$, maka $\sqrt{a^2} = a$.

Ingat bahwa bila $a \in \mathbb{R}$, maka $|a|$ didefinisikan sebagai

$$|a| \stackrel{\text{def}}{=} \begin{cases} a, & \text{bila } a \geq 0 \\ -a, & \text{bila } a < 0 \end{cases}$$

PERSAMAAN dan PERTIDAKSAMAAN NILAI MUTLAK

Bila $x, a \in \mathbb{R}$ berlaku

a. $|x| = a \Longleftrightarrow x = -a$ atau $x = a$

b. $|x| \leq a \Longleftrightarrow -a \leq x \leq a$

c. $|x| \geq a \Longleftrightarrow x \leq -a$ atau $x \geq a$.

Contoh

1. $|2x+5| = 3 \Longleftrightarrow 2x+5 = 3$ atau $2x+5 = -3$. Jadi $x = \frac{3-5}{2} = -1$ atau $x = \frac{-3-5}{2} = -4$
2. $|3x-2| \leq 1$ didapat
$$\begin{aligned} |3x-2| \leq 1 &\Longleftrightarrow -1 \leq 3x-2 \leq 1 \\ &\Longleftrightarrow -1+2 \leq 3x-2+2 \leq 1+2 \\ &\Longleftrightarrow 1 \leq 3x \leq 3 \\ &\Longleftrightarrow \frac{1}{3} \leq x \leq 1. \end{aligned}$$
3. $|5 - \frac{2}{x}| < 3$ didapat
$$\begin{aligned} |5 - \frac{2}{x}| < 3 &\Longleftrightarrow -3 < 5 - \frac{2}{x} < 3 \\ &\Longleftrightarrow -3-5 < -5+5-\frac{2}{x} < -5+3 \\ &\Longleftrightarrow -8 < -\frac{2}{x} < -2 \\ &\Longleftrightarrow 4 > \frac{1}{x} > 1 \\ &\Longleftrightarrow \frac{1}{4} < x < 1. \end{aligned}$$

Sifat 4,5 dan Contoh 2, 3 dikerjakan dalam SAGEMATH Shell sebagai berikut.

In [3]:
```
a=SR.var('a',domain='real') #mendefinisikan a peubah real
t=bool(sqrt(a^2) == abs(a))
html("Bila $a\\in\\mathbb{R}$, apakah $\\sqrt{a^2}=|a|$?\
$\\color{red}{%s}$"%latex(t))
```

Out[3]: Bila $a \in \mathbb{R}$, apakah $\sqrt{a^2} = |a|$?$\color{red}{\mathrm{True}}$

In [4]:
```
# diskusikan hasil berikut
assume(a > 0)
t=bool(sqrt(a^2) == a)
html("Bila $a>0$, apakah $\\sqrt{a^2}=a$? $\\color{red}{%s}$"%latex(t))
```

Out[4]: Bila $a > 0$, apakah $\sqrt{a^2} = a$? $\color{red}{\mathrm{True}}$

Tetapi bila $a \in \mathbb{R}$, maka $\sqrt{a^2} = a$ tidak benar. Hal ini bisa ditunjukkan dalam SageMath Shell sebagai berikut:

In [5]:
```
forget(a>0)
t=bool(sqrt(a^2) == a)
html("Bila $a\\in\\mathbb{R}$, apakah $\\sqrt{a^2}=a$?\
$\\color{red}{%s}$"%latex(t))
```

Out[5]: Bila $a \in \mathbb{R}$, apakah $\sqrt{a^2} = a$?$\color{red}{\mathrm{False}}$

In [6]:

```
# 2. Pertaksamaan : |3x−2|≤1

p=plot(abs(3*x-2),(x,0,2),ymax=5,legend_label="$f(x)=\
\\ |3x-2|$",thickness=3,color="red")
p1=plot(abs(3*x-2),(x,1/3,1),ymax=5,thickness=3,color="red",fill ='axis')
p2=arrow((0.9,0.5), (1.1,0.5))
p3=text("Daerah Penyelesaian",(1.56,0.5))
sl=solve(abs(3*x-2)<=1,x)
(p+p1+p2+p3).show(figsize=4,gridlines='True')
pretty_print(html("Penyelesaian dari $|3x-2|\\leq 1$\
adalah: $%s$"%latex(sl)))
```



Penyelesaian dari $|3x-2| \leq 1$adalah: $\left[\left[x = 1\right], \left[x = \left(\frac{1}{3}\right)\right], \left[\left(\frac{1}{3}\right) < x, x < 1\right]\right]$

In [7]:

```
# 3. Pertakaksamaan : |5-2/x|<3

a = plot([])
a +=line([(1/4,0),(1/4,3)],linestyle=":",thickness=3)
a +=line([(1,0),(1,3)],linestyle=":",thickness=3)
a +=point([(1/4,3),(1,3)],pointsize=40)

p=plot(abs(5-2/x),(x,1/10,2),ymax=5,ymin=-1,legend_label="$f(x)=\
\\ |5-\\frac{2}{x}|$",\
       thickness=3,color="green")
p1=plot(abs(5-2/x),(x,1/4,1),ymax=5,thickness=3,\
        color="green",fill ='axis')
p2=text("Daerah Penyelesaian",(1.56,0.5))
p3=arrow((0.9,0.5), (1.1,0.5))
sl=solve(abs((5*x-2)/x)<3,x)
(p+p1+p2+p3+a).show(figsize=4,frame='True',gridlines='True')
pretty_print(html("Penyelesaian dari $|5-2/x|<3$ adalah:\
$%s$"%latex(sl)))
```



Penyelesaian dari $|5-2/x| < 3$ adalah:
$\left[\left[\left(\frac{1}{4}\right) < x, x < \left(\frac{2}{5}\right)\right], \left[x = \left(\frac{2}{5}\right)\right], \left[\left(\frac{2}{5}\right) < x, x < 1\right]\right]$

Menggambar grafik suatu fungsi

Pada bagian ini, dibahas bagaimana menggambar grafik menggunakan SageMath dan bagaimana memanfaatkan berbagai plot pilihannya. Dibahas secara rinci beberapa pilihan yang akan berguna dalam pembahasan kalkulus. Sintaks dasar untuk menggambar grafik fungsi $y = f(x)$ dengan $a \leq x \leq b$ adalah plot (f, x, a, b).

Contoh 1
Lakukan fungsi berikut

$$f(x) = \begin{cases} \tan\left(\frac{\pi x}{4}\right), & \text{bila } |x| < 1, \\ x, & \text{bila } |x| \geq 1 \end{cases}$$

kedalam sel SageMath. Hal ini bisa dkerjakan sebagai berikut, lalu gambar grafiknya.

In [8]:

```
def f(x):
    if abs(x)<1:
        return tan(pi*x/4)
    else:
        return x

plot(f,(x,-2,2),figsize=4,gridlines='True')
```

Out[8]:



Contoh 2

Gambarkan dua garfik fungsi $f(x) = 2x^2 - 4x + 2$ dan $g(x) = 25\sin(3x)$ dalam satu sumbu yang sama sehingga terlihat titik potong kedua grafik tersebut,

Jawab : Lakukakan dalam Sel **SageMath** sebagai berikut:

In [9]:

```
P = plot(sin, (0,10))
P.show()
```



In [10]:
```
g=plot((2*x^2-4*x+2,25*sin(3*x)),x,-3,5,figsize=4,thickness=3,\
       legend_label=("$2x^2-4x+2$","$25\\, \\sin(3x)$"))
g.show()
```



Contoh 3

Gambar grafik fungsi $f(x) = \dfrac{2x^2 + x + 2}{x + 1}$ dan $g(x) = \dfrac{\cos(2x)}{2}$ pada sumbu yang sama. Selanjutnya diskusikan hasilnya.

Jawab : Lakukan dalam Sel **SageMath** sebagai berikut:

In [11]:
```
g=plot(((2*x^2+x+2)/(x+1),cos(2*x)/2),x,-5,5,figsize=4,thickness=2)
g.show()
```



Terlihat bahwa hasil gambar grafik kedua fungsi tidak bisa dibedakan. Dalam hal ini gambar grafik fungsi $g(x) = \dfrac{\cos(2x)}{2}$ tidak tampak begitu jelas sebab nilai $g(x)$ diantara -1 dan $+1$ terlalu kecil dibandingkan dengan nilai-nilai dari $f(x) = \dfrac{2x^2 + x + 2}{x + 1}$. Untuk itu digunakan perintah `ymin` dan `ymax` sebagaimana berikut.

In [12]:

```
g=plot(((x^2+x+2)/(x+1),cos(2*x)/2),x,-5,5,ymin=-10,ymax=6,figsize=4,\
       thickness=2,legend_label=("$(x^2+x+2)/(x+1)$","$\\cos(2x)/2)$"))
g.show(frame='True',gridlines='minor')
```



Kembali ke File Utama (http://localhost:8888/notebooks/Workshop_Utama_Unsyiah_2020.ipynb)

Kalkulus

Limit

Untuk menghitung limit fungsi $f(x)$ bila x mendekati a dalam Sel SageMath digunakan perintah limit(f(x),x=a). Untuk menghitung limit fungsi $f(x)$ bila x mendekati a dari kiri dalam hal ini berarti $x < a$ menggunakan perintah **limit(f(x),x=a,dir='minus')**. Untuk menghitung limit fungsi $f(x)$ bila x mendekati a dari kanan dalam hal ini berarti $a < x$ menggunakan perintah **limit(f(x),x=a,dir='plus')**.

Contoh 1

Tentukan $\lim\limits_{x\to 1} \dfrac{2x^2+x+3}{x+2}$.

Jawab:

Sebelum menghitung nilai limit, diberikan tabel perhitungan nilai dari $\dfrac{2x^2+x+3}{x+2}$ bila x mendekati 1 dari kiri ataupun dari kanan sebagai berikut:

In [1]:
```
A=[(1-(10)^(-a)).n(digits=6) for a in [0..5]]
B=[(1+10^(-b)).n(digits=6) for b in [5,4,3,2,1,0]]
C=A+B
html(table([(t,(2*t^2+t+3)/(t+2)) for t in C],\
           header_row=["$x\\quad$  ", "  $\\dfrac{2x^2+x+3}{x+2}$"],\
           frame=True))
```

Out[1]:

x	$\dfrac{2x^2+x+3}{x+2}$
0.000000	1.50000
0.900000	1.90345
0.990000	1.99003
0.999000	1.99900
0.999900	1.99990
0.999990	1.99999
1.00001	2.00001
1.00010	2.00010
1.00100	2.00100
1.01000	2.01003
1.10000	2.10323
2.00000	3.25000

Dari hasil tabel diatas terlihat bahwa nilai dari $\dfrac{2x^2+x+3}{x+2}$ bila x mendekati 1 dari kiri ataupun dari kanan mendekati 2. Dengan demikian $\lim_{x \to 1} \frac{2x^2 + x + 3}{x + 2} = 2$. Hal ini bisa diselidiki dalam Sel **SageMath** sebagaimana berikut:

In [2]:
```
#  limit f(x) untuk x mendekati 1
var('x')

limit((2*x^2+x+3)/(x+2),x=1)
```

Out[2]: 2

In [3]:
```
#  limit f(x) untuk x mendekati 1 dari kiri

limit((2*x^2+x+3)/(x+2),x=1,dir='minus')
```

Out[3]: 2

In [4]:
```
#  limit f(x) untuk x mendekati 1 dari kanan

limit((2*x^2+x+3)/(x+2),x=1,dir='plus')
```

Out[4]: 2

Contoh 2

Tentukan $\lim_{x \to 1} \frac{x^2 - x}{x - 1}$.

Jawab :

Dibuat tabel nilai dari $\frac{x^2 - x}{x - 1}$ bila x mendekati 1 baik dari kiri atau kanan sebagai berikut:

In [5]:
```
A=[(1-(10)^(-a)).n(digits=6) for a in [0..5]]
B=[(1+10^(-b)).n(digits=6) for b in [5,4,3,2,1,0]]
C=A+B
html(table([(t,(t^2-t)/(t-1)) for t in C],\
           header_row=["$x\\quad$", "  $\\dfrac{x^2-x}{x-1}$"],\
           frame=True))
```

Out[5]:

x	$\dfrac{x^2-x}{x-1}$
0.000000	−0.000000
0.900000	0.900000
0.990000	0.989998
0.999000	0.998987
0.999900	1.00000
0.999990	1.00000
1.00001	1.00000
1.00010	1.00000
1.00100	1.00095
1.01000	1.01000
1.10000	1.10000
2.00000	2.00000

Dari hasil tabel diatas terlihat bahwa nilai dari $\dfrac{x^2-x}{x-1}$ bila x mendekati 1 dari kiri ataupun dari kanan adalah 1. Dengan demikian $\lim\limits_{x\to 1}\dfrac{x^2-1}{x-1}=1$. Hal ini bisa diselidiki dalam Sel SageMath sebagaimana berikut:

In [6]:
```
#  limit f(x) untuk x mendekati 1
var('x')

limit((x^2-x)/(x-1),x=1)
```

Out[6]: 1

In [7]:
```
#  limit f(x) untuk x mendekati 1 dari kiri
var('x')

limit((x^2-x)/(x-1),x=1,dir='minus')
```

Out[7]: 1

In [8]:
```
#  limit f(x) untuk x mendekati 1 dari kanan
var('x')

limit((x^2-x)/(x-1),x=1,dir='plus')
```

Out[8]: 1

Berikut ini adalah gambar dari grafik $y = \dfrac{x^2 - x}{x - 1}$.

In [9]:
```
p1 = plot((x^2 - x)/(x - 1), x, -4,4,legend_label=\
          '$y=\\frac{x^2-x}{x-1}$',thickness=2)
p2 = point((1,1), rgbcolor='red', pointsize=70, faceted=True)
g = p1+p2
g.show(xmin=-4, xmax=4, ymin=-4, ymax=4,figsize=4,frame=\
       'True',gridlines='True')
```



Fungsi Kontinu

Fungsi bernilai real $f(x)$ dikatakan kontinu di $x = a$ bila $\lim\limits_{x \to a} f(x) = f(a)$ dimana hasil limit tidak tergantung pada arah pendekatan x mendekati a dari kiri ataupun dari kanan.

Fungsi bernilai real yang didefinisikan oleh $f(x) \overset{\text{def}}{=} \begin{cases} 1, & \text{bila } x \geq 0 \\ 0, & \text{bila } x < 0 \end{cases}$, diskontinu di $x = 0$. Diskusikan hal ini mengapa!

Catatan, dalam **Contoh 2** $f(x) = \dfrac{x^2 - x}{x - 1}$ jelas tidak terdefinisi di $x = 1$, tetapi $\lim\limits_{x \to 1} \dfrac{x^2 - 1}{x - 1} = 1$. Lalu bagaimana bila didefinisikan $f(1) \overset{\text{def}}{=} 1$. Hasil ini bisa diringkas dalam teorema berikut.

Teorema
Bila fungsi bernilai real $f(x)$ takterdefinisi di $x = a$ dan $\lim\limits_{x \to a} f(x) = A$, maka fungsi $f(x)$ akan kontinu di $x = a$ asalkan diasumsikan $f(a) \overset{\text{def}}{=} A$.

Berdasarkan teorema diatas, maka fungsi yang didefinisikan oleh

$$f(x) \overset{\text{def}}{=} \begin{cases} \dfrac{x^2 - x}{x - 1}, & \text{bila } x \neq 1 \\ 1, & \text{bila } x = 1 \end{cases}$$

kontinu di $x = 1$, Sebab $f(1) = 1 = \lim\limits_{x \to 1} f(x)$.

Contoh 1
Selidiki apakah fungsi bernilai real yang diberikan oleh $f(x) = \dfrac{x}{|x|}$ kontinu di $x = 0$.

Jawab :

Karena untuk x mendekati 0 dari kanan nilai $\lim\limits_{x \to 0^+} \dfrac{x}{|x|} = \dfrac{x}{x} = 1$ dan untuk x mendekati 0 dari kiri nilai $\lim\limits_{x \to 0^-} \dfrac{x}{|x|} = \dfrac{x}{-x} = -1$, maka $\lim\limits_{x \to 0} \dfrac{x}{|x|}$ tidak ada. Dengan demikian berdasarkan teorema yang telah dibahas $f(x)$ diskontinu di $x = 0$. Gambar berikut menjelaskan bahwa di $x = 0$ pada $f(x) = \dfrac{x}{|x|}$ terjadi loncatan.

In [10]:
```
g=plot(x/abs(x),x,-5,5,figsize=4,thickness=2,legend_label=\
        '$f(x)=\\dfrac{x}{|x|}$')
g.show()
```



Contoh 2
Perkirakan nilai dari $f(x) = \cos\left(\dfrac{\pi}{x}\right)$ disekitar x mendekati 0.

Jawab :
Digambar dulu grafik dari $f(x) = \cos\left(\dfrac{\pi}{x}\right)$ dalam Sel SageMath sebagai berikut:

In [11]:
```
# Gambar cos(pi/x) dan perkiran nilainya bila x mendekati nol

var('x')
teks=text("$f(x)=\\cos(\\dfrac{\\pi}{x})$",(2.4,0.9),\
          fontsize =13,color="green")
p=plot(cos(pi/x),(x,-3,3),color="green")
(p+teks).show(figsize=4)
```



Selanjutnya dibuat tabel nilai dari $f(x) = \cos\left(\dfrac{\pi}{x}\right)$ disekitar x mendekati nol sebagai berikut:

In [12]:
```
A=[(-(10)^(-a)).n(digits=4) for a in [0..5]]
B=[(10^(-b)).n(digits=4) for b in [5,4,3,2,1,0]]
C=A+B
html(table([(x,cos(pi/x).n(digits=4)) for x in C],\
           header_row=["$x\\quad$",\
                       "$\\cos\\left(\\dfrac{\\pi}{x}\\right)$"],\
           frame=True))
```

Out[12]:

x	$\cos\left(\dfrac{\pi}{x}\right)$
-1.000	-1.000
-0.1000	1.000
-0.01000	1.000
-0.001000	1.000
-0.0001000	0.9973
-0.00001000	0.7421
0.00001000	0.7421
0.0001000	0.9973
0.001000	1.000
0.01000	1.000
0.1000	1.000
1.000	-1.000

Dari tabel tampak bahwa bila x mendekati nol baik dari kiri ataupun dari kanan nilai dari $\cos\left(\dfrac{\pi}{x}\right)$ mendekati satu. Tetapi dari gambar grafik $\cos\left(\dfrac{\pi}{x}\right)$ tampak bahwa untuk $x \to 0$, nilai $\cos\left(\dfrac{\pi}{x}\right)$ berisolasi dari $+1$ ke -1.

Misalnya untuk nilai $a \neq b$ dengan
$$a = \pm\frac{1}{10000000000000001} \quad \text{dan} \quad b = \pm\frac{1}{10000000000000002}$$
didapat
$$\cos\left(\frac{\pi}{a}\right) \; = -1 \quad \text{dan} \quad \cos\left(\frac{\pi}{b}\right) \; = 1.$$
Diselidiki hal ini menggunakan Sel SageMath sebagai berikut.

In [13]:

```
a=1/100000000000000000001
b=1/100000000000000000002
pretty_print(html("$a=%s$"%a))
pretty_print(html("$b=%s$"%b))
pretty_print(html("$\\cos(\\pi/a)=%s$"%cos(pi/a)))
pretty_print(html("$\\cos(-\\pi/a)=%s$"%cos(-pi/a)))
pretty_print(html("$\\cos(\\pi/b)=%s$"%cos(pi/b)))
pretty_print(html("$\\cos(-\\pi/b)=%s$"%cos(-pi/b)))
```

$a = 1/100000000000000000001$

$b = 1/100000000000000000002$

$\cos(\pi/a) = -1$

$\cos(-\pi/a) = -1$

$\cos(\pi/b) = 1$

$\cos(-\pi/b) = 1$

In [14]:

```
var('x')
teks=text("$f(x)=\\cos(\\dfrac{\\pi}{x})$",(0.4,0.9),\
          fontsize =13,color="green")
p=plot(cos(pi/x),(x,-0.3,0.3),color="green")
(p+teks).show(figsize=4)
```



Dari gambar grafik $f(x) = \cos\left(\dfrac{\pi}{x}\right)$ tampak sekali bahwa terjadi osilasi yang rapat sekali dari nilai $+1$ ke -1. Dengan demikian nilai $\lim\limits_{x\to 0} \cos\left(\dfrac{\pi}{x}\right)$ tidak ada.

In [15]:

```
limit(cos(pi/x),x=0)
```

Out[15]: ind

disini **ind** artinya "***indefinite but bounded***". Selanjutnya bagaimana bila sekarang $f(x) = x\cos\left(\dfrac{\pi}{x}\right)$

In [16]:

```
x = var('x')
f = x*cos(pi/x)
L=f.limit(x = 0)

pretty_print(html("$\\lim\\limits_{x\\to 0}%s=\
%s$"%(latex(f),latex(L))))
```

$$\lim_{x\to 0} x\cos\left(\frac{\pi}{x}\right) = 0$$

Mengapa $\lim\limits_{x\to 0} x\cos\left(\frac{\pi}{x}\right) = 0$? Hal ini bisa diamati perilaku grafik dari $f(x) = x\cos\left(\frac{\pi}{x}\right)$ disekitar niali x mendekati 0.

In [17]:

```
var('x')
teks=text("$f(x)=x\\cos(\\frac{\\pi}{x})$",(0.25,0.5),fontsize=\
          15,color="green")
p=plot(x*cos(pi/x),(x,-0.5,0.5),ymin=-0.5,ymax=0.5,color=\
       "green",thickness=2)
(p+teks).show(figsize=4)
```



Terlihat disekitar nilai x mendekati 0 perilaku grafik $f(x) = x\cos\left(\frac{\pi}{x}\right)$ tetap berisolasi dari nilai positip ke negatif tetapi semakin menuju ke nilai 0. Dengan demikian jelas bahwa $\lim\limits_{x\to 0} x\cos\left(\frac{\pi}{x}\right) = 0$. Penggambaran grafik suatu fungsi bisa memperkirakan perilaku grafik fungsi tersebut disekitar titik pengamatan. Misalnya gambar grafik dari $f(x) = e^{-\frac{1}{x}}$ disekitar x mendekati 0 yang diberikan oleh gambar berikut:

In [18]:
```
plot(e^(-1/x),(x,-3,3),ymin=-.1,ymax=20,figsize=4,thickness=2,\
     legend_label="$y=e^{-1/x}$")
```

Out[18]:



Terlihat bahwa grafik $f(x) = e^{-\frac{1}{x}}$ diskontinu di $x = 0$. Disekitar x mendekati 0 ada dua hal yaitu bila x mendekati 0^- maka nilai $f(x)$ menuju ke $+\infty$ sedangkan bila x mendekati 0^+, maka $f(x)$ menuju 0. Hal ini bisa diseldiki dalam Sel SageMath sebagai berikut:

In [19]:
```
x = var('x')
f = e^(-1/x)
L1=f.limit(x = 0,dir='-')
L2=f.limit(x = 0,dir='+')

pretty_print(html("$\\lim\\limits_{x\\to 0^{-}}%s=\
%s$"%(latex(f),latex(L1))))
pretty_print(html("$\\lim\\limits_{x\\to 0^{+}}%s=\
%s$"%(latex(f),latex(L2))))
```

$$\lim_{x\to 0^-} e^{\left(-\frac{1}{x}\right)} = +\infty$$

$$\lim_{x\to 0^+} e^{\left(-\frac{1}{x}\right)} = 0$$

Beberapa sifat dasar limit dari fungsi : Bila $\lim\limits_{x\to a} f(x) = A$ dan $\lim\limits_{x\to a} g(x) = B$

1. $\lim\limits_{x\to a} f(x) \pm g(x) = A \pm B$
2. $\lim\limits_{x\to a} f(x)g(x) = AB$

3. $\lim_{x \to a} \frac{f(x)}{g(x)} = \frac{A}{B},\ B \neq 0.$

Pada bagian berikutnya ditunjukkan bahwa $\lim_{x \to 0} \frac{\sin(x)}{x} = 1$. Untuk hal ini perhatikan gambar lingkaran pusat di O dengan jari-jari sama dengan 1 sebagai berikut.



Bila nilai x adalah panjang busur AM. Sehingga didapat $2MP < 2AM < 2MT$ atau $\sin(x) < x < \tan(x)$ atau $1 < \frac{x}{\sin(x)} < \frac{1}{\cos(x)}$. Dengan demikian bila x mendekati 0 didapat $1 < \lim_{x \to 0} \frac{x}{\sin(x)} < 1$. Jadi $\lim_{x \to 0} \frac{x}{\sin(x)} = 1$. Selanjutnya dengan menggunakan sifat dasar limit bagian 3 didapat $\lim_{x \to 0} \frac{\sin(x)}{x} = \lim_{x \to 0} \frac{1}{\frac{x}{\sin(x)}} = 1/1 = 1$.

Apa yang telah dibahas ini dapat diselidiki dalam Sel SageMath sebagai berikut:

In [20]:
```
A=[(-(10)^(-a)).n(digits=4) for a in [0..5]]
B=[(10^(-b)).n(digits=4) for b in [5,4,3,2,1,0]]
C=A+B
html(table([(x,sin(x)/x) for x in C],\
           header_row=["$x\\qquad$",\
                       "$\\dfrac{\\sin(x)}{x}$"],\
           frame=True))
```

Out[20]:

x	$\dfrac{\sin(x)}{x}$
-1.000	0.8415
-0.1000	0.9983
-0.01000	1.000
-0.001000	1.000
-0.0001000	1.000
-0.00001000	1.000
0.00001000	1.000
0.0001000	1.000
0.001000	1.000
0.01000	1.000
0.1000	0.9983
1.000	0.8415

Terlihat dari tabel perhitungan nilai-nilai dari $\dfrac{\sin(x)}{x}$ untuk x mendekati 0 baik dari kiri ataupun kanan adalah 1.

In [21]:
```
# Gambar sin(x)/x

var('x')
pt1 = point((0,1), rgbcolor="red", pointsize=50, faceted=True)
p=plot(sin(x)/x,(x,-20,20),ymax=1.2,frame=True,legend_label=\
       "$\\frac{\\sin(x)}{x}$")
(p+pt1).show(figsize=4,gridlines='True')
```



In [22]:
```
# Menghitung nilai limit dari sin(x)/x untuk x mendekati 0

l=limit(sin(x)/x,x=0)
html("$\\lim\\limits_{x\\to 0}\\dfrac{\\sin(x)}{x}\\ \\ = %s$"%l)
```

Out[22]: $\lim\limits_{x\to 0}\dfrac{\sin(x)}{x}\ \ = 1$

In [23]:
```
var('theta')
A=limit((1-cos(theta))/(theta)^2,theta=0)
html("$\\lim\\limits_{\\theta\\to 0}\
\\dfrac{1-\\cos(\\theta)}{\\theta^2}=%s$"%latex(A))
```

Out[23]: $\lim\limits_{\theta\to 0}\dfrac{1-\cos(\theta)}{\theta^2}=\frac{1}{2}$

Dari hasil $\lim\limits_{\theta\to 0}\dfrac{1-\cos(\theta)}{\theta^2}=\frac{1}{2}$, maka untuk nilai θ cukup sangat kecil sekali didapat $\cos(\theta)\approx 1-\dfrac{1}{2}\theta^2$. Hasil pendekatan ini akurat untuk beberapa digit desimal bila $|\theta| < 1/4$.

Limit mendekati takhingga.

Hitung $\lim\limits_{x\to+\infty}\dfrac{1}{x}$.

Jawab:

Hal ini dikerjakan dalam Sel SageMath sebagai berikut:

In [24]:

```
var('x')
A=limit(1/x,x=oo,dir='minus')
B=limit(1/x,x=oo,dir='plus')
C = bool(A==B)
D = limit(1/x,x=oo)

pretty_print(html("Apakah $\\lim\\limits_{x\\to\
+\\infty^{-}}\\dfrac{1}{x}=\\lim\\limits_{x\\to\
+\\infty^{+}}\\dfrac{1}{x}$ ? $%s$"%latex(C)))
pretty_print(html("$\\lim\\limits_{x\\to\
+\\infty}\\dfrac{1}{x}=%s$"%latex(D)))
```

Apakah $\lim\limits_{x\to +\infty^{-}}\dfrac{1}{x}=\lim\limits_{x\to +\infty^{+}}\dfrac{1}{x}$? True

$\lim\limits_{x\to +\infty}\dfrac{1}{x}=0$

In [25]:

```
# Contoh limit mendekati takhingga

A=limit((x^2+3)/(2*x^2+x+1),x=oo)
html("$\\lim\\limits_{x\\to +\\infty}\
\\dfrac{x^2+3}{2x^2+x+1}=%s$"%latex(A))
```

Out[25]: $\lim\limits_{x\to +\infty}\dfrac{x^2+3}{2x^2+x+1}=\frac{1}{2}$

Menghitung limit dari $\left(1+\dfrac{1}{x}\right)^x$ untuk x mendekati $+\infty$ dan $(1+x)^{\frac{1}{x}}$ untuk x mendekati 0.

In [26]:

```
x = var('x')
f = (1+1/x)^x
L=f.limit(x = oo)

html("$\\lim\\limits_{x\\to +\\infty}%s=\
%s$"%(latex(f),latex(L)))
```

Out[26]: $\lim\limits_{x\to +\infty}\left(\frac{1}{x}+1\right)^x=e$

In [27]:

```
x = var('x')
f = (1+x)^(1/x)
L=f.limit(x = 0)

html("$\\lim\\limits_{x\\to 0}%s=\
%s$"%(latex(f),latex(L)))
```

Out[27]: $\lim\limits_{x\to 0}(x+1)^{\left(\frac{1}{x}\right)}=e$

Deferensial fungsi satu peubah

Diberikan fungsi $y = f(x)$, asumsikan x sebarang tetap, maka $\Delta y \overset{\text{def}}{=} f(x + \Delta x) - f(x)$. Sehingga didapat $\dfrac{\Delta y}{\Delta x} = \dfrac{f(x + \Delta x) - f(x)}{\Delta x}$. Bila Δx mendekati 0, maka didapat

$$\lim_{\Delta x \to 0} \frac{\Delta y}{\Delta x} = \lim_{\Delta x \to 0} \frac{f(x + \Delta x) - f(x)}{\Delta x}.$$

Bila $\lim\limits_{\Delta x \to 0} \dfrac{\Delta y}{\Delta x}$ ada, maka $\lim\limits_{\Delta x \to 0} \dfrac{\Delta y}{\Delta x}$ dinamakan diferensial dari fungsi f di x, dalam hal ini dinotasikan sebagai $\dfrac{dy}{dx}$. Jadi

$$\frac{dy}{dx} = \lim_{\Delta x \to 0} \frac{f(x + \Delta x) - f(x)}{\Delta x}.$$

Contoh

Tentukan $\dfrac{dy}{dx}$, bila $y = f(x) = x^2$.

Jawab

$$\begin{aligned}
\frac{dy}{dx} &= \lim_{\Delta x \to 0} \frac{f(x + \Delta x) - f(x)}{\Delta x} \\
&= \lim_{\Delta x \to 0} \frac{(x + \Delta x)^2 - x^2}{\Delta x} \\
&= \lim_{\Delta x \to 0} \frac{(x^2 + 2x\Delta x + \Delta x^2) - x^2}{\Delta x} \\
&= \lim_{\Delta x \to 0} \frac{2x\Delta x + \Delta x^2}{\Delta x} \\
&= \lim_{\Delta x \to 0} (2x + \Delta x) \\
&= 2x.
\end{aligned}$$

Hal ini dilakukan dalam Sel SageMath sebagai berikut:

In [28]:
```
x,h=var('x h')
f(x)=x^2
Delta_y=f(x+h)-f(x)
Dyh=(Delta_y/h).expand()
Dl=limit(Dyh,h=0)
pretty_print(html("$y=f(x)=%s$"%latex(f(x))))
pretty_print(html("$\\dfrac{dy}{dx}=\
\\lim\\limits_{\\Delta x\\to 0}\
\\dfrac{\\Delta y}{\\Delta x}=\
\\lim\\limits_{\\Delta x\\to 0}\
\\dfrac{f(x+\\Delta x)-f(x)}{\\Delta x}=\
%s$"%latex(Dl)))
```

$y = f(x) = x^2$

$$\frac{dy}{dx} = \lim_{\Delta x \to 0} \frac{\Delta y}{\Delta x} = \lim_{\Delta x \to 0} \frac{f(x + \Delta x) - f(x)}{\Delta x} = 2\,x$$

In [29]:
```
# Cara Langsung

A=diff(f(x),x)
pretty_print(html("$y=f(x)=%s$"%latex(f(x))))
pretty_print(html("$\\dfrac{dy}{dx}=%s$"%latex(A)))
```

$y = f(x) = x^2$

$\dfrac{dy}{dx} = 2\,x$

In [31]:
```
x,h=var('x h')
f(x)=x^3+2*x^2+3*x+5
Delta_y=f(x+h)-f(x)
Dyh=(Delta_y/h).expand()
Dl=limit(Dyh,h=0)
pretty_print(html("$y=f(x)=%s$"%latex(Dl)))
```

$y = f(x) = 3\,x^2 + 4\,x + 3$

In [32]:
```
# Cara Langsung

A=diff(f(x),x)
pretty_print(html("$y=f(x)=%s$"%latex(f(x))))
pretty_print(html("$\\dfrac{dy}{dx}=%s$"%latex(A)))
```

$y = f(x) = x^3 + 2\,x^2 + 3\,x + 5$

$\dfrac{dy}{dx} = 3\,x^2 + 4\,x + 3$

In [33]:
```
x=var('x')
f(x)=3/(x^2-1)
A=diff(f(x),x)
pretty_print(html("$y=f(x)=%s$"%latex(f(x))))
pretty_print(html("$\\dfrac{dy}{dx}=%s$"%latex(A)))
```

$y = f(x) = \frac{3}{x^2-1}$

$\dfrac{dy}{dx} = -\frac{6\,x}{\left(x^2-1\right)^2}$

Limit Menggunakan Aturan **L'Hospital**, contoh berikut menghitung limit menggunakan aturan L'Hospital yang dilakukan dalam Sel **SageMath**.

In [35]:
```
%display latex

var('x')
f(x)=ln(x)
g(x)=1/x
f(x)/g(x)
```

Out[35]: $x \log(x)$

In [36]:
```
lim(f(x)/g(x),x=0,dir='plus')
```

Out[36]: 0

In [37]:
```
diff(f(x),x)/diff(g(x),x)
```

Out[37]: $-x$

In [38]:
```
lim(diff(f(x),x)/diff(g(x),x),x=0,dir='plus')
```

Out[38]: 0

Dapatkan persamaan garis singgung pada $f(x) = x + \sin(2\pi x)$ di $(1, 1)$.

In [39]:
```
x=var('x')
f(x)=x+sin(2*pi*x);
df(x)=diff(f(x),x);
y(x)=df(1)*(x-1)+1;
h=y(x).simplify_full()

pretty_print(html("$f(x)=%s$"%latex(f(x))))
pretty_print(html("Garis singgung pada $f(x)$ di titik $(1,1)$\
adalah :  $\\ \\ y=%s$"%latex(h)))
```

$f(x) = x + \sin(2\,\pi x)$

Garis singgung pada $f(x)$ di titik $(1, 1)$adalah : $\ \ y = -2\,\pi + (2\,\pi + 1)x$

In [40]:
```
p1=plot(f(x), x, 0,2, color='blue',legend_label=\
        "$x+\\sin(2\\pi x)$");
p2=plot(y(x), x, 0,2, color='red',legend_label=\
        "$(2\\pi +1)x-2\\pi$");
p3 = point((1,1), pointsize=30, rgbcolor='green')
show(p1+p2+p3, ymax=3, ymin=-3,figsize=5,gridlines='True')
```



In [41]:
```
x,k,w = var('x k w')
f = x^3 * e^(k*x) * sin(w*x)
pretty_print(html("$$f(x) = %s$$"%latex(f)))
pretty_print(html("$$\\dfrac{df(x)}{dx}=\
%s$$"%latex(f.diff(x))))
```

$$f(x) = x^3 e^{(kx)} \sin(wx)$$

$$\frac{df(x)}{dx} = wx^3 \cos(wx)e^{(kx)} + kx^3 e^{(kx)} \sin(wx) + 3\,x^2 e^{(kx)} \sin(wx)$$

Kita juga dapat membedakan dan mengintegrasikan menggunakan perintah berikut:

In [42]:
```
R = PolynomialRing(QQ,"x")
x = R.gen()
p = x^2 + 1
show(p.derivative())
show(p.integral())
```

$2x$

$\frac{1}{3}x^3 + x$

Atau dengan perintah beikut:

In [43]:
```
R = PolynomialRing(QQ,"x")
x = R.gen()
p = x^2 + 1
show(p)
show(p.derivative())
```

$x^2 + 1$

$2x$

Titik Kritis

Kita dapat menemukan titik kritis dari fungsi *piecewise* yang ditentukan sebagai berikut:

In [44]:
```
x = PolynomialRing(RationalField(), 'x').gen()
f1 = x^0
f2 = 1-x
f3 = 2*x
f4 = 10*x-x^2
f = piecewise([((0,1),f1), ((1,2),f2), ((2,3),f3), ((3,10),f4)])
show(f.critical_points())
```

$[5.0]$

Deret Pangkat

Sage menawarkan beberapa cara untuk membangun dan bekerja dengan deret pangkat.

Untuk mendapatkan deret Taylor dari ekspresi fungsi gunakan perintah `.taylor()` pada ekspresi:

In [45]:
```
f0,k,x = var('f0 k x')
g = f0/sinh(k*x)^4
h = g.taylor(x,0,7)
pretty_print(html("$$g(x)=%s$$"%latex(g)))
pretty_print(html("Deret Taylor dari $$g(x),\\ x,0,7: %s$$"%latex(h)))
```

$$g(x) = \frac{f_0}{\sinh(kx)^4}$$

Deret Taylor dari

$$g(x),\ x, 0, 7 : -\frac{62}{22275} f_0 k^6 x^6 + \frac{41}{2835} f_0 k^4 x^4 - \frac{62}{945} f_0 k^2 x^2 + \frac{11}{45} f_0 - \frac{2 f_0}{3 k^2 x^2} + \frac{f}{k^4}$$

Ekspansi deret pangkat formal dari fungsi dapat diperoleh dengan perintah `.series()` :

In [46]:
```
f =1/(2-cos(x))
g = f.series(x,7)
html("Ekspansi Deret Pangkat Formal$$f(x)=%s:\
%s$$"%(latex(f),latex(g)))
```

Out[46]: Ekspansi Deret Pangkat Formal

$$f(x) = -\frac{1}{\cos(x) - 2} : 1 + (-\frac{1}{2})x^2 + \frac{7}{24}x^4 + (-\frac{121}{720})x^6 + \mathcal{O}\left(x^7\right)$$

Manipulasi tertentu pada deret semacam itu sulit dilakukan saat ini. Ada dua alternatif: gunakan subsistem Maxima dari SageMath untuk fungsionalitas simbolik penuh:

In [47]:
```
f = log(sin(x)/x)
g = f.taylor(x,0,10)
h = maxima(f).powerseries(x,0)._sage_()
pretty_print(html("Deret Taylor dari $$f(x) = %s,\\ x,0,10 :\
%s$$"%(latex(f),latex(g))))
pretty_print(html("Deret (menggunakan Maxima) dari $$f(x) = %s :\\ \
%s$$"%(latex(f),latex(h))))
```

Deret Taylor dari

$$f(x) = \log\left(\frac{\sin(x)}{x}\right),\ x, 0, 10 : -\frac{1}{467775} x^{10} - \frac{1}{37800} x^8 - \frac{1}{2835} x^6 - \frac{1}{180} x^4 - \frac{1}{6} x^2$$

Deret (menggunakan Maxima) dari

$$f(x) = \log\left(\frac{\sin(x)}{x}\right) : \sum_{i_2=1}^{+\infty} \frac{2^{2 i_2 - 1} (-1)^{i_2} x^{2 i_2} \mathrm{bern}(2 i_2)}{i_2 (2 i_2)!}$$

Atau kita dapat menggunakan ring deret pangkat formal untuk komputasi cepat. Di sisi lain, hal ini adalah fungsi simbolis yang hilang:

In [48]:

```
R.<w> = QQ[[]]
ps = w + 17/2*w^2 + 15/4*w^4 + O(w^6)
a = (1+ps).log()
b = (ps^1000).coefficients()
pretty_print(html("$$p(w) = %s$$"%latex(ps)))
pretty_print(html("$$\\log(1+p(w)) = %s$$"%latex(a)))
pretty_print(html("Koefien $$p(w)^{1000} = %s$$"%latex(b)))
```

$$p(w) = w + \frac{17}{2}w^2 + \frac{15}{4}w^4 + O(w^6)$$

$$\log(1+p(w)) = w + 8w^2 - \frac{49}{6}w^3 - \frac{193}{8}w^4 + \frac{301}{5}w^5 + O(w^6)$$

Koefien

$$p(w)^{1000} = \left[1, 8500, 36088875, 102047312625, \frac{1729600092867375}{8}\right]$$

[Kembali ke File Utama (http://localhost:8888/notebooks/Workshop_Utama_Unsyiah_2020.ipynb)](http://localhost:8888/notebooks/Workshop_Utama_Unsyiah_2020.ipynb)

In [ ]:

```

```

INTEGRAL

Pada bagian ini dibahas integral dalam SageMath. Perintah **integrate(f(x),x)** memberikan integral tak tentu (atau antiderivatif) dari f terhadap x. Perintah integrate dapat mengevaluasi semua fungsi rasional dan sejumlah fungsi transendental, termasuk fungsi eksponensial, logaritmik, trigonometrik, dan invers trigonometri. Untuk mengintegral fungsi, f (x,y) terhadap x perintahnya **integrate(f(x,y),x)**.

Hitung integral berikut:

$$\int_1^4 (x^2 + 2x - 1)dx$$

Jawab :

In [1]:
```
x=var('x')
f=x^2+2*x-1
g=integrate(f,x)
h=integrate(f,x,1,4)
p1=plot(f,(x,0,5), legend_label='$x^2+2x-1$',figsize=[4,4],thickness=3)
p2=plot(f,(x,1,4), figsize=[4,4],fill = 'axis')
(p1+p2).show(figsize=4,gridlines=True)
pretty_print(html("Nilai dari: $\\int\\limits_{1}^{4}(%s)dx=\
\\left.%s\\right|^{4}_{1}=%s$"%(latex(f),latex(g),latex(h))))
```



Nilai dari: $\int\limits_{1}^{4}(x^2 + 2\,x - 1)dx = \left.\frac{1}{3}\,x^3 + x^2 - x\right|^{4}_{1} = 33$

Hitung integral : $\int \frac{2x}{\sqrt{x+1}}dx$.

Jawab:

In [2]:
```
x=var('x')
f=(2*x)/sqrt(x+1)
g=integrate(f,x)
html("Nilai dari: $$\\int\\left(%s\\right)\\,\
dx=%s+c$$"%(latex(f),latex(g)))
```

Out[2]: Nilai dari:

$$\int\left(\frac{2\,x}{\sqrt{x+1}}\right)\,dx=\frac{4}{3}\,(x+1)^{\frac{3}{2}}-4\,\sqrt{x+1}+c$$

Hitung integral : $\int 3x^2\cos(x^3)\,dx$.

Jawab:

In [3]:
```
x=var('x')
f=(3*x^2)*cos(x^3)
g=integrate(f,x)
html("Nilai dari: $$\\int\\left(%s\\right)\\,\
dx=%s+c$$"%(latex(f),latex(g)))
```

Out[3]: Nilai dari:

$$\int\left(3\,x^{2}\cos\left(x^{3}\right)\right)\,dx=\sin\left(x^{3}\right)+c$$

Hitung integral : $\int \frac{2x^5+x^2+x+1}{x^2-1}\,dx$.

Jawab:

In [4]:
```
x=var('x')
f=(2*x^5+x^2+x+1)/(x^2-1)
g=integrate(f,x)
html("Nilai dari: $$\\int\\left(%s\\right)\\,\
dx=%s+c$$"%(latex(f),latex(g)))
```

Out[4]: Nilai dari:

$$\int\left(\frac{2\,x^{5}+x^{2}+x+1}{x^{2}-1}\right)\,dx=\frac{1}{2}\,x^{4}+x^{2}+x+\frac{1}{2}\,\log\left(x+1\right)+\frac{5}{2}\,\log\left(x-1\right)+c$$

Hitung integral : $\int \frac{x^4+2x^3+3x+1}{(x^2+1)^2}\,dx$. Diskusikan menggunakan substitusi teknik integral apa!

Jawab:

In [5]:
```
x=var('x')
f=(x^4+2*x^3+3*x+1)/(x^2+1)^2
g=integrate(f,x)
html("Nilai dari: $$\\int\\left(%s\\right)\\,\
dx=%s+c$$"%(latex(f),latex(g)))
```

Out[5]: Nilai dari:

$$\int\left(\frac{x^4+2\,x^3+3\,x+1}{\left(x^2+1\right)^2}\right)\,dx=x+\frac{2\,x-1}{2\,\left(x^2+1\right)}-\arctan(x)+\log\left(x^2+1\right)+c$$

Hitung luas daerah yang dibatasi oleh kurva $f(x)=2x^3-8x^2+4x$ dan $g(x)=x^2$

Jawab:
Pertama ditentukan dulu titik potong kedua kurva $f(x)=2x^3-8x^2+2x$ dan $g(x)=x^2$ dengan **SageMath** sebagai berikut:

In [6]:
```
x=var('x')
f=2*x^3-8*x^2+4*x
g=x^2
solve(f(x)==g(x),x)
```

Out[6]:
```
[x == 4, x == (1/2), x == 0]
```

Terlihat dua kurva $f(x)=2x^3-8x^2+4x$ dan $g=x^2$ berpotongan di $x=0,\frac{1}{2}$ dan $x=4$.
Senjutnya kedua kurva kita gambar dalam SageMath sebgai berikut:

In [7]:
```
h=plot((f,g),x,-0.1,4.1,figsize =[4,4],ymin=-10,ymax=17,\
       fill='axis',legend_label=('$f(x)=2x^3-8x^2+4x$','$g(x)=x^2$'))
h.show(gridlines=True)
```



Dari plot Gambar terlihat bahwa $f(x) \geq g(x),\ 0 \leq x \leq \frac{1}{2}$ dan $g(x) \geq f(x),\ 0 \leq x \leq 4$. Dengan demikian luas daerah yang dibatasi oleh kurva $f(x)$ dan $g(x)$ adalah:

$$\int\limits_{0}^{\frac{1}{2}}(f(x)-g(x))dx+\int\limits_{\frac{1}{2}}^{4}(g(x)-f(x))dx.$$

Luas ini dihitung dengan **SageMath** sebagai berikut:

In [8]:
```
a=integrate(f-g,x,0,1/2)
b=integrate(g-f,x,1/2,4)
c=a+b
pretty_print(html("Luas daerah yang dibatasi oleh\
$f(x)=%s$ dan $g(x)=%s$ adalah:"%(latex(f),latex(g))))
pretty_print(html("$$\\int\\limits_{0}^{\\frac{1}{2}}\
(f(x)-g(x))dx+\\int\\limits_{\\frac{1}{2}}^{4}(g(x)-f(x))dx\
=%s+%s=%s$$"%(latex(a),latex(b),latex(c))))
```

Luas daerah yang dibatasi oleh$f(x) = 2\,x^3 - 8\,x^2 + 4\,x$ dan $g(x) = x^2$ adalah:

$$\int\limits_{0}^{\frac{1}{2}}(f(x)-g(x))dx+\int\limits_{\frac{1}{2}}^{4}(g(x)-f(x))dx=\frac{5}{32}+\frac{1029}{32}=\frac{517}{16}$$

Integral Tak-tentu dari $f(x)$ yaitu $\int f(x)\,dx$ telah dibuat secara interaktif menggunakan GeoGebra

Klasik versi : 6.0.596.0-offline (24 Juli 2020) sebagai Kalkulator Integral Tak-Tentu dapat diakses di: https://www.geogebra.org/m/vbjkrb3d (https://www.geogebra.org/m/vbjkrb3d)

Berikut ini adalah program interaktif untuk menghitung integral tak-tentu dari suatu fungsi f dalam **SageMath**:

In [9]:

```
x=var('x')
@interact
def _(f = input_box(default = x^3-2*x^2+3,width=25),auto_update=False):
    q=integrate(f,x)
    pretty_print(html("Nilai integral:\
    $$\\int\\left(%s\\right)\\,dx=%s+c$$"%(latex(f),latex(q))))
```

A Jupyter widget could not be displayed because the widget state could not be found. This could happen if the kernel storing the widget is no longer available, or if the widget state was not saved in the notebook. You may be able to create the widget by running the appropriate cells.

Sebelum dibuat program interaktif dari integral tertentu dari fungsi f pada interval tutup $[a, b]$ ditampilkan kalkulator dari integral tertentu menggunakan **GeoGebra** program bisa diakses di : https://www.geogebra.org/m/syvvvvxj (https://www.geogebra.org/m/syvvvvxj)

Berikutnya diberikan program interaktif untuk menghitung integral tertentu dari suatu fungsi f pada interval $[a, b]$. Output dari program adalah nilai eksak dan nilai pendekatan numeriknya.

In [10]:

```
x=var('x')
@interact
def _(f = e^(-x^2),a=-4,b=4,auto_update=False):
    q=integrate(f,x)
    p=integrate(f,x,a,b)
    r=plot([f],x,a,b,gridlines='minor',\
           legend_label='automatic',fill = true)
    r.show(figsize=4)
    pretty_print(html("Nilai integral:\
    $$\\int\\limits_{%s}^{%s}\\left(%s\\right)\\,dx=\
    \\left.%s\\right|_{%s}^{%s}=%s\
    \\approx %s$$"%(latex(a),latex(b),latex(f),latex(q),\
                    latex(a),latex(b),latex(p),latex(n(p)))))
```

A Jupyter widget could not be displayed because the widget state could not be found. This could happen if the kernel storing the widget is no longer available, or if the widget state was not saved in the notebook. You may be able to create the widget by running the appropriate cells.

Jumlahan Reimann dan Integral Tertentu

Ulasan Jumlahan Riemann: Partisi dari interval tertutup $[a, b]$ adalah himpunan titik $P = \{x_0, x_1, \ldots, x_n\}$ di $[a, b]$ sedemikian rupa sehingga

$$a = x_0 < x_1 < \cdots < x_n = b.$$

Diberikan fungsi f yang didefinisikan pada interval tertutup $[a, b]$ dan partisi $P = \{x_0, x_1, \dots, x_n\}$ di $[a, b]$, ingat bahwa Jumlahan Riemann f pada $[a, b]$ relatif terhadap partisi P adalah jumlah dari formula berikut:

$$\sum_{i=0}^{n} f(x_i^*)\Delta x_i,$$

dimana $\Delta x_i = x_i - x_{i-1}$ dan x_i^* adalah sebarang titik di subinterval $[x_{i-1}, x_i]$ diasumsikan bahwa $\Delta x_i = \frac{b-a}{n}$ untuk semua i. Dengan demikian jumlahan Reimann itu perkiraan dari luas daerah antara kurva f dan sumbu-x sepanjang interval tutup $[a, b]$. Luas yang tepat diberikan oleh integral tertentu dari f pada $[a, b]$ didefinisikan sebagai limit dari jumlahan Reimann untuk $n \to \infty$ dan dinotasikan oleh $\int_a^b f(x)dx$, jadi

$$\int_a^b f(x)dx \stackrel{\text{def}}{=} \lim_{n\to\infty} \sum_{i=0}^{n} f(x_i^*)\Delta x_i.$$

Integral tertentu ini ada asalkan limit dari jumlahan Reimann ada. Untuk suatu fungsi kontinu f di $[a, b]$, dapat ditunjukkan bahwa integral tertentu $\int_a^b f(x)dx$ ada.

Approksimasi Integral

Implementasi teknik integrasi numerik diberikan berikut ini: Jumlah Riemann kiri, jumlah Riemann kanan, Peraturan Titik Tengah, Aturan Trapesium, dan Aturan Simpson. Ubah dan evaluasi kode SageMath sesuai keinginan. Setiap fungsi mengambil input fungsi f pada interval tutup $[a, b]$, dan bilangan bulat positip n sebagai banyaknya partisi. Ingat $\Delta x = \frac{b-a}{n}$ dan $x_i = a + i\Delta x$ untuk setiap $0 \leq i \leq n$.

Jumlahan Reimann Kiri

Fungsi Reimann_kiri menampilkan perkiraan jumlah Riemann sisi kiri Fungsi Reimann_kiri menampilkan perkiraan jumlah Riemann sisi kiri $\int_a^b f(x)dx$ menggunakan sebanyak n partisi interval:

$$\int_a^b f(x)dx \approx \sum_{i=0}^{n} f(x_{i-1})\Delta x = \Delta x(f(x_0) + f(x_1) + \cdots + f(x_{n-1})).$$

Berikut ini fungsi Reimann_kiri dalam **SageMath**:

In [11]:

```
def Reimann_kiri(fcn,a,b,n):
    # # output: Reimann-kiri approksimasi jumlahan Reimann\
    ## dari int_a^b fcn(x)dx menggunakan n langkah
    Deltax = (b-a)*1.0/n
    return Deltax*sum([fcn(a+Deltax*i) for i in range(n)])

#Contoh berikut hasil eksaknya 33
n=10000
a=1
b=4
f(x)=x^2+2*x-1
print(Reimann_kiri(f,a,b,n).n())
```

```
32.9968500450000
```

Jumlahan Reimann Kanan

Fungsi Reimann_kanan menampilkan perkiraan jumlah Riemann sisi kanan Fungsi Reimann_kanan menampilkan perkiraan jumlah Riemann sisi kanan $\int_a^b f(x)dx$ menggunakan sebanyak n partisi interval:

$$\int_a^b f(x)dx \approx \sum_{i=1}^{n} f(x_i)\Delta x = \Delta x(f(x_1) + f(x_2) + \cdots + f(x_n)).$$

Berikut ini fungsi Reimann_kanan dalam **SageMath**:

In [12]:

```
def Reimann_kanan(fcn,a,b,n):
# output: Reimann-kanan approksimasi jumlahan Reimann dari\
# int_a^b fcn(x)dx menggunakan n langkah
    Deltax = (b-a)*1.0/n
    return Deltax*sum([fcn(a+Deltax*(i+1)) for i in range(n)])

#Contoh jawaban eksak 33
n=10000
a=1
b=4
f(x)=x^2+2*x-1
print(Reimann_kanan(f,a,b,n).n());
```

```
33.0031500450000
```

Aturan Titik Tengah

Fungsi Reimann_tengah menampilkan perkiraan jumlah Riemann sisi kanan Fungsi Reimann_tengah menampilkan perkiraan jumlah Riemann $\int_a^b f(x)dx$ menggunakan sebanyak n partisi interval:

$$\int_a^b f(x)dx \approx \sum_{i=1}^{n} f(\bar{x}_{i-1})\Delta x,$$

dimana $\bar{x}_i$ adalah titik tengah dari interval tutup $[x_{i-1}, x_i]$ yaitu $\bar{x}_i = \frac{x_i+x_{i-1}}{2}$. Berikut ini fungsi Reimann_tengah dalam **SageMath**:

In [13]:
```
def Reimann_tengah(fcn,a,b,n):
# output: aturan titik tengah approksimasi dari\
# int_a^b fcn(x)dx menggunakan n langkah
    Deltax = (b-a)*1.0/n
    xs=[a+Deltax*i for i in range(n+1)]
    ysmid=[fcn((xs[i]+xs[i+1])/2) for i in range(n)]
    return Deltax*sum(ysmid)

#Contoh jawaban eksak 33
n=10000;
a=1
b=4
f(x)=x^2+2*x-1
print(Reimann_tengah(f,a,b,n).n())
```

```
32.9999999775001
```

Aturan Trapesium

Fungsi Reimann_trapesium menampilkan perkiraan jumlah Riemann $\int_a^b f(x)dx$ menggunakan sebanyak n partisi interval:

$$\int_a^b f(x)dx \approx \frac{\Delta x}{2}(f(x_0) + 2f(x_1) + 2f(x_2) + \cdots + 2f(x_{n-1} + f(x_n)).$$

Berikut ini fungsi Reimann_trapesium dalam **SageMath**:

In [14]:
```
def Reimann_trapesium(fcn,a,b,n):
# output: aturan trapesoidal approksimasi dari\
# int_a^b fcn(x)dx menggunakan n langkah
    Deltax = (b-a)*1.0/n
    coeffs = [2]*(n-1)
    coeffs = [1]+coeffs+[1]
    valsf = [fcn(a+Deltax*i) for i in range(n+1)]
    return (Deltax/2)*sum([coeffs[i]*valsf[i] for i in range(n+1)])

#Contoh jawaban eksak 33
n=10000;
a=1
b=4
f(x)=x^2+2*x-1
print(Reimann_trapesium(f,a,b,n).n())
```

```
33.0000000450000
```

Aturan Simpson

Fungsi Reimann_simpson menampilkan perkiraan jumlah Riemann $\int_a^b f(x)dx$ menggunakan sebanyak n (n harus genap) partisi interval:

$$\int_a^b f(x)dx \approx \frac{\Delta x}{3}(f(x_0) + 4f(x_1) + 2f(x_2) + 4f(x_3) + \cdots + 2f(x_{n-2} + 4f(x_{n-1}) + f(x_n)).$$

Berikut ini fungsi Reimann_simpson dalam **SageMath**:

In [15]:
```
def Reimann_simpson(fcn,a,b,n):
# output: aturan simpson  approksimasi dari\
# int_a^b fcn(x)dx menggunakan n langkah (n harus genap)
    Deltax = (b-a)*1.0/n
    n2=int(n/2)
    coeffs = [4,2]*n2
    coeffs = [1] +coeffs[:n-1]+[1]
    valsf = [fcn(a+Deltax*i) for i in range(n+1)]
    return (Deltax/3)*sum([coeffs[i]*valsf[i] for i in range(n+1)])

#Contoh jawaban eksak 33
n=10000;
a=1
b=4
f(x)=x^2+2*x-1
print(Reimann_simpson(f,a,b,n).n())
```

```
33.0000000000001
```

Volume benda putar

Diberikan suatu kurva $y = f(x)$ dan $x \in [a, b]$. Bila benda dibatatasi oleh $f(x)$ dan $x = a, x = b$ diputar pada sumbu-x, maka vulume benda putar adalah:

$$V = \pi \int_a^b (f(x))^2 dx.$$

Contoh

Diberikan benda yang dibatasi oleh $f(x) = (x-1)^2 + 1$ dan $x = -1, x = 3$ diputar pada sumbu-x, maka volume benda putar adalah:

$$
\begin{aligned}
V &= \pi \int_{-1}^{3} \left((x-1)^2+1\right)^2 dx \\
&= \pi \int_{-1}^{3} \left((x-1)^4+2(x-1)^2+1\right) dx \\
&= \pi \left[\frac{1}{5}(x-1)^5+\frac{2}{3}(x-1)^3+x)\right]\Bigg|_{-1}^{3} \\
&= \pi \left[\left(\frac{32}{5}+\frac{16}{3}+3\right)-\left(-\frac{32}{5}-\frac{16}{3}-1\right)\right] \\
&= \frac{412\pi}{15}.
\end{aligned}
$$

Kita selesaikan contoh ini dengan menggunakan **SageMath** sebagai berikut:

In [16]:

```
u=var('u')
f=(u-1)^2+1
plot(f,u,-1,3,figsize=3,fill=True)
```

Out[16]:



In [17]:

```
r=revolution_plot3d(f,(u,-1,3),show_cuve=True,opacity=7,\
                    parallel_axis='x',aspect_ratio=(1,1,1))
r.show()
```



In [18]:

```
int=pi*integrate(f^2,u,-1,3)
g=(x-1)^2+1
h=pi*integrate(g^2+1,x)
pretty_print(html("$V=\\pi\\int\\limits_{-1}^{3}\
\\left(%s\\right)^2 dx=\\left. %s\\right|_{-1}^{3}=%s$"\
                  %(latex(g),latex(h),latex(int))))
```

$$V = \pi \int\limits_{-1}^{3} \left((x-1)^2+1\right)^2 dx = \frac{1}{15}\,\pi\left(3\,x^5 - 15\,x^4 + 40\,x^3 - 60\,x^2 + 75\,x\right)\Big|_{-1}^{3} = \frac{412}{15}\,\pi$$

Panjang kurva:

Panjang kurva $y = f(x)$ untuk $x = a$ dan $x = b$ adalah S, dengan $dS^2 = dx^2 + dy^2 = \left[1 + \left(\frac{dy}{dx}\right)^2\right] dx^2$, maka $dS = \sqrt{1 + \left(\frac{dy}{dx}\right)^2}\,dx$. Jadi panjang S adalah:

$$S = \int\limits_a^b \sqrt{1 + \left(\frac{dy}{dx}\right)^2}\,dx.$$

Contoh

Hitung panjang kurva $y = (x-1)^2 + 1$ dari $x = -1$ ke $x = 3$. Jawab :

$$\begin{aligned} S &= \int\limits_{-1}^{3} \sqrt{1 + 4(x-1)^2}\,dx \\ &= \left(\frac{1}{2}\sqrt{4x^2 - 8x + 5}\,x - \frac{1}{2}\sqrt{4x^2 - 8x + 5} + \frac{1}{4} arsinh(2x-2)\right)\Big|_{-1}^{3} \\ &= 2\sqrt{17} + \frac{1}{2}\,arsinh(4). \end{aligned}$$

In [19]:

```
y=(x-1)^2+1
dy=diff(y,x)
t = integrate(sqrt(1+dy^2),x)
s=integrate(sqrt(1+dy^2),x,-1,3)
pretty_print(html("Panjang kurva :\
$$S=\\int\\limits_{-1}^{3}\\sqrt{1+%s}dx=\
\\left. %s\\right|_{-1}^{3}=%s$$"%(latex(dy^2),latex(t),latex(s))))
```

Panjang kurva :

$$S = \int\limits_{-1}^{3} \sqrt{1 + 4\,(x-1)^2}\,dx = \frac{1}{2}\,\sqrt{4\,x^2 - 8\,x + 5}x - \frac{1}{2}\,\sqrt{4\,x^2 - 8\,x + 5} + \frac{1}{4}\,\mathrm{arsinh}(2\,x$$

Luas Permukaan Benda Putar

Diberikan suatu kurva $y = f(x)$ dan $x \in [a, b]$. Bila benda dibatatasi oleh $f(x)$ dan $x = a, x = b$ diputar pada sumbu-x, maka luas permukaan benda putar adalah A dengan $dA = 2\pi\ y\ ds$ sehingga didapat:

$$A = 2\pi \int\limits_a^b f(x)\sqrt{1 + \left(f'(x)\right)^2}dx.$$

Contoh

Diberikan kurva $y = f(x) = (x-1)^2 + 1$ dan $x \in [a, b]$. Bila benda dibatatasi oleh $f(x)$ dan $x = -11, x = 3$ diputar pada sumbu-x, maka luas permukaan benda putar adalah A dengan $dA = 2\pi\ y\ ds$ sehingga didapat:

$$A = 2\pi \int\limits_{-1}^3 f(x)\sqrt{1 + \left(f'(x)\right)^2}dx.$$

Hasil hitungan diberikan dalam sel SageMath berikut:

In [20]:
```
var('x y')
y=(x-1)^2+1
dy=diff(y,x)
t = integrate(y*sqrt(1+dy^2),x)
A = integrate(y*sqrt(1+dy^2),x,-1,3)
pretty_print(html("$f(x)=%s$"%latex(y)))
pretty_print(html("$A=2\\pi\\int\\limits_{-1}^{3}f(x)\
\\sqrt{1+f'(x)^2}\\ dx=\\left.\\left(%s\\right)\\right|_{-1}^{3}$"\
                    %latex(t)))
pretty_print(html("$\\ \\ \\ =%s$"%latex(A)))
```

$$f(x) = (x-1)^2 + 1$$

$$A = 2\pi \int\limits_{-1}^{3} f(x)\sqrt{1 + f'(x)^2}\ dx$$

$$= \left(\frac{1}{16}\left(4x^2 - 8x + 5\right)^{\frac{3}{2}}x - \frac{1}{16}\left(4x^2 - 8x + 5\right)^{\frac{3}{2}} + \frac{15}{32}\sqrt{4x^2 - 8x + 5}x - \frac{15}{32}\sqrt{4x^2 \cdot\right.$$

$$= \frac{49}{8}\sqrt{17} + \frac{15}{32}\,\mathrm{arsinh}(4)$$

Kembali ke File Utama (http://localhost:8888/notebooks/Workshop_Utama_Unsyiah_2020.ipynb)

In [ ]:

Metoda Newton Raphson

Metode Newton adalah teknik untuk mendekati akar persamaan bentuk $f(x) = 0$.Yang diperlukan adalah perkiraan awal untuk akar, yang disebut x_1 , dan formula berulang:

$$x_{n+1} = x_n - \frac{f(x_n)}{f'(x_n)} \qquad (1)$$

$x_1, x_2, x_3, \ldots$ diharapkan konvergen ke akar yang diinginkan.

Persamaan (1) diturunakn dari persamaan garis melaui titik $(x_0, f(x_0))$ dengan gradien $m = f'(x_0)$ yaitu

$$y - f(x_0) = f'(x_0)(x - x_0). \qquad (2)$$

Garis pada Persamaan (2) memotong sumbu-x sehigga didapat:

$$x - x_0 = -\frac{f(x_0)}{f'(x_0)} \Longrightarrow x = x_0 - \frac{f(x_0)}{f'(x_0)}. \qquad (3)$$

Iterasi Persamaan (3) akan didapat Persamaan (1). Sebelum dibuat program untuk metoda Newton Raphson, diberikan suatu program interaktif mengenai ide dari metoda Newton Raphson.

In [1]:

```
@interact
def _(f=log(x),a=0,b=6,s=slider(1,10,0.1),auto_update=false):
    dy = diff(f,x)
    g=dy(s)*x+(f(s)-s*dy(s))

    p1 = plot(f,(x,a,b),figsize=5)
    p2 = plot(g,(x,-b,b))
    p3 = point((s,f(s)),size=50,color='red')
    p4 = point((s,0),size=50,color='red')
    p5 = point(((-f(s)+s*dy(s))/dy(s),0),size=50,color='red')
    p6 = line([(s,0),(s,f(s))],color='red')
    p7 = line([(s,0), ((-f(s)+s*dy(s))/dy(s),0)],color='red',thickness=2)
    (p1+p2+p3+p4+p5+p6+p7).show()
```

f log(x)

a 0

b 6

s 1.00

Run Interact



Implementasi Metode Newton ditunjukkan pada blok kode di SelSageMathi.Kita dapat mengubah fungsi f, dan tebakan awal x_1. Mengklik evaluasi akan menjalankan satu iterasi dari metode Newton dan mengembalikan dua perkiraan berikutnya. Kita jangan ragu untuk menambahkan beberapa baris lagi kode untuk menemukan lebih banyak iterasi, cukup hanya dengan duplikat dua baris kode terakhir dan perbarui indeks:

x6 = IterasiNewton (x5);
print(x5);

Silakan proses program berikut (disini fungsi f adalah polinomial):

In [2]:

```
import math
import numpy as np
from sympy import *

x = Symbol('x');
y = x^3+5*x-5;
dy = y.diff(x);
f = lambdify(x, y, 'numpy')
df = lambdify(x, dy, 'numpy')
def IterasiNewton(x):
    return x-(f(x)/df(x));

x1=1.;
print(x1);
x2=IterasiNewton(x1);
print(x2);
x3=IterasiNewton(x2);
print(x3);
x4=IterasiNewton(x3);
print(x4);
x5=IterasiNewton(x4);
print(x5);
```

```
1.00000000000000
0.875000000000000
0.868843683083512
0.868830020408451
0.868830020341475
```

Program berikut adalah modifikasi dari program sebelumnya sebelumnya.

In [3]:
```
import math
import numpy as np
from sympy import *

x = var('x');
y = x^3+5*x-5;
dy = y.diff(x);
f = lambdify(x, y, 'numpy')
df = lambdify(x, dy, 'numpy')
def IterasiNewton(x):
    return x-(f(x)/df(x));

xn = 1.;
print(xn);
for i in range(10):
    xn=IterasiNewton(xn)
    print('{:.20f}'.format(xn))
```

```
1.00000000000000
0.87500000000000000000
0.86884368308351200000
0.86883002040845100000
0.86883002034147500000
0.86883002034147500000
0.86883002034147500000
0.86883002034147500000
0.86883002034147500000
0.86883002034147500000
0.86883002034147500000
```

Sebelum mengakhiri pembahasan Metoda Newton-Raphson, kita ingatkan lagi beberapa hal. Diberikan fungsi f yang kontinu differesiabel di $[a, b]$, misalkan $x_f \in [a, b]$ dengan $f(x_f) = 0$ dan suatu nilai awal $x_0 \in [a, b]$. Beberapa hal kita diskusikan melalui pertanyaan berikut:

1. Apakah nilai awal sebarang $x_0 \in [a, b]$ menjamin bahwa untuk beberapa $n \in \{1, 2, 3, \dots\}$ menjamin bahwa $x_n \to x_f$?
2. Apakah selalu bila nilai awal x_0 cekup dekat dengan x_f, maka dijamin $x_n \to x_f$ untuk beberapa $n \in \{1, 2, 3, \dots\}$?
3. Tetapi sebaliknya bila x_0 csemakin jauh dengan x_f, maka tidak mungkin $x_n \to x_f$ untuk suatu $n \in \{1, 2, 3, \dots\}$. Untuk menjawab hal ini lihat di website: [https://www.geogebra.org/m/pefb23xn (https://www.geogebra.org/m/pefb23xn)](https://www.geogebra.org/m/pefb23xn), disini pembhasan ditulis menggunakan perangkat lunak GeoGebra Klasik versi : 6.0.593.0-offline (07 Juli 2020).

Program berikut adalah program interaktif dari metode Newton, disini input fungsi f tidak harus polinomial bisa juga fungsi logaritma

[Program Metode Newton (http://localhost:8888/notebooks/Interaktif_Newton2020.ipynb)](http://localhost:8888/notebooks/Interaktif_Newton2020.ipynb)

[Kembali ke File Utama (http://localhost:8888/notebooks/Workshop_Utama_Unsyiah_2020.ipynb)](http://localhost:8888/notebooks/Workshop_Utama_Unsyiah_2020.ipynb)

In [ ]:

```

```

Program berikut adalah program interaktif dari metode Newton, disini input fungsi f tidak harus polinomial bisa juga fungsi logaritma.

In [1]:

```
@interact
def _(f=log(x)+x^2,xn=input_box(1),n=input_box(5),auto_update=False):
    df=diff(f,x)
    for i in range(n):
        xn=N(xn-(f(xn)/df(xn)))
        pretty_print(html("$x_{%s}=%s$"%(latex(i+1),latex(xn))))
```

f	x^2 + log(x)
xn	1
n	5

Run Interact

$x_1 = 0.666666666666667$

$x_2 = 0.652909253842097$

$x_3 = 0.652918640413836$

$x_4 = 0.652918640419205$

$x_5 = 0.652918640419205$

Kembali ke File Utama (http://localhost:8888/notebooks/Workshop_Utama_Unsyiah_2020.ipynb)

In [ ]:

Bilangan "Floating Point"

Bagaimana kita bisa mewakili bilangan real di komputer? Secara umum, bilangan-bilngan ini tidak dapat dikodekan dengan jumlah informasi yang terbatas, dan dengan demikian bilangan-bilangan itu tidak dapat direpresentasikan secara tepat. Diperlukan perkiraan mereka menggunakan jumlah memori yang terbatas. Suatu standar telah muncul di sekitar perkiraan bilangan real dengan jumlah informasi yang terbatas: representasi floating-point.

Dalam bagian ini, kita akan menemukan: deskripsi dasar bilangan *floating-point* dan berbagai jenis bilangan-bilangan ini tersedia di SageMath, dan ditunjukkan beberapa sifat mereka. Contohnya akan menunjukkan beberapa kesulitan yang kita temui ketika komputasi dengan bilangan *floating-point* dan beberapa trik untuk mengatasinya. Diharapkan bahwa pembaca akan mengembangkan pendekatan yang diperlukan hati-hati. Untuk menyimpulkan, kita akan mencoba menggambarkan beberapa sifat yang harus dipenuhi oleh metode numerik ketika kita menggunakan bilangan-bilangan ini.

Dalam Sage seperti pada MPFR *Multiple Precision Floating-Point Reliable*), bilangan *floating-point* presisi p berbentuk $sm2^{e-p}$ dengan $s \in \{-1, 1\}$, $2^{p-1} \leq m < 2^p$ dan $-2^B + 1 \leq e \leq 2^B - 1$, dimana $B = 30$ pada sistem 32-bit dan $B = 62$ pada sistem 64-bit. Selain itu, ada nilai-nilai khusus $+0$, -0, +infinity, -infinity dan NaN (yang merupakan singkatan dari Not-a-Number).

Operasi dalam bahasan ini yang merupakan pembungkus langsung fungsi MPFR "dibulatkan dengan benar"; dijelaskan secara singkat apa artinya ini. Asumsikan bahwa kita dapat melakukan operasi dengan tepat, pada bilangan real, untuk mendapatkan hasil r. Jika hasil ini dapat direpresentasikan sebagai suatu bilangan *floating-point*, maka akan dihasilkan bilangan itu.

Kalau tidak, hasilnya r adalah antara dua bilangan *floating-point*. Untuk mode pembulatan terarah (dibulatkan ke **plus takhingga**, dibulatkan ke **minus takhingga**, dibulat ke **nol**), dihasilkan bilangan *floating-point* dalam arah yang ditunjukkan dari r. Untuk pembulatan ke terdekat, akan dihasilkan bilangan *floating-point* yang terdekat dengan r.

Ini membuat satu kasus tidak ditentukan: dalam pembulatan ke mode terdekat, apa yang terjadi jika r tepat setengah dari dua bilangan *floating-point* terdekat? Dalam hal itu, kita membulatkan bilangan ke bilangan dengan *mantissa* genap (mantissa adalah bilangan m dalam representasi di atas).

Pertimbangkan sekumpulan bilangan *floating-point* dengan presisi p. (Di sini diidentifikasi $+0$ dan -0, dan mengabaikan NaN.) Dapat ditunjukkan bahwa ada korespondensi satu-satu pada antara bilangan-bilangan folating-point dengan segmen bilangan bulat, di mana 0 dipetakan ke 0 dan bilangan *floating-point* yang berdekatan dipetakan ke bilangan bulat yang berdekatan. Bilangan bulat yang sesuai dengan bilangan *floating-point* yang diberikan dinamakan "**rank *floating-point***" dari bilangan tersebut (ini bukan terminologi standar).

In [1]:
```
1/2 in QQ
```

Out[1]:
```
True
```

In [2]:
```
# System bilangan real maximum

RR(sys.maxsize)
```

Out[2]: 9.22337203685478e18

Sage menyediakan dua jenis bilangan *floating-point*:

1. bilangan "**presisi ganda**" sebagai mana dijelaskan bilangan *floating-point* sebelumnya: bilangan-bilangan ini disediakan oleh prosesor komputer; di SageMath, ia termasuk dalam kelas **RDF**:

In [3]:
```
xrdf = RDF(11.173)
xrdf
```

Out[3]: 11.173

2. bilangan *floating-point* dengan presisi sebarang: setiap kelas **RealField** - atau **Reals** - mendefinisikan satu himpunan bilangan floating-point dengan presisi yang diberikan (dan mungkin dengan mode pembulatan yang diberikan. Misalnya, untuk mendeklarasikan **x100=3/7** dengan presisi pada 100 digit biner, dalam SageMath ditulis:

In [4]:
```
R100 = RealField(100)  # presisi: 100 bits.
x100 = R100(3/7)
x100
```

Out[4]: 0.42857142857142857142857142857

Bilangan dalam himpunan **RealField(p)** mempunyai bentuk: $(-1)^s\, 0.d_1\, d_2\, \cdots\, d_p\; 2^e$, dengan $s \in \{0, 1\}$: signifikan mempunyai p digit biner dan e mungkin memiliki 30 digit biner (atau lebih pada beberapa komputer). Tersedia presisi implisit:

In [5]:
```
Rdefault = RealField() # presisi baku  53 bits
xdefault = Rdefault(3/7)
xdefault
```

Out[5]: 0.428571428571429

dan dimungkinkan untuk memeriksa presisi semua bilangan **floating-point** menggunakan perintah **prec()**:

In [6]:
```
x = 3.7

type(x),x.prec()
```

Out[6]: (<class 'sage.rings.real_mpfr.RealLiteral'>, 53)

Di sini, **real_mpfr.RealLiteral** berarti bahwa himpunan bilangan x dimana ia berada diimplementasikan oleh pustaka (library) **GNU MPFR**. Diingat kembali, bahwa jenis variabel secara otomatis ditentukan oleh sisi kanan dalam tugas (assignment):

In [7]:
```
x = 1.0             # x di RealField()
x = 0.1e+1          # sama: x di RealField()
x = 1               # x di ZZ, himpunan bilangan bulat
x = RDF(1)          # x adalah  bilangan "mesin" presisi-ganda
x = RDF(1.)         # sama: x adalah  bilangan "mesin" presisi-ganda
x = RDF(0.1e+1)     # idem
x = 4/3             # x adalah bilangan rasional di QQ
R = RealField(20)
x = R(1)            # x adalah bilangan floating-point 20-bit
```

Berikut ini kita berikan elemen-elemen x berada sesuai dengan himpunannya dan kita uji kebenarannya.

In [8]:
```
R17d = RealField(17,rnd='RNDZ')
R17d
```

Out[8]:
```
Real Field with 17 bits of precision and rounding RNDZ
```

In [9]:
```
x = R17d(4.37)
a = x.exact_rational()
pretty_print(html("$x = %s$"%latex(x)))
pretty_print(html("$a = %s$"%latex(a)))
pretty_print(html("Apakah $x = %s\\in\\mathbb{Z}?\
\\ %s$"%(latex(x),latex(x in ZZ))))
pretty_print(html("Apakah $a = %s\\in\\mathbb{Z}?\
\\ %s$"%(latex(a),latex(x in ZZ))))
pretty_print(html("Apakah $x = %s\\in\\mathbb{Q}?\
\\ %s$"%(latex(x),latex(x in QQ))))
pretty_print(html("Apakah $a = %s\\in\\mathbb{Q}?\
\\ %s$"%(latex(a),latex(x in QQ))))
pretty_print(html("Apakah $x = %s\\in\\mathrm{RealField()}?\
\\ %s$"%(latex(x),latex(x in RealField()))))
pretty_print(html("Apakah $a = %s\\in\\mathrm{RealField()}?\
\\ %s$"%(latex(a),latex(x in RealField()))))
pretty_print(html("Apakah $x = %s\\in\\mathrm{R17d}?\
\\ %s$"%(latex(x),latex(x in R17d))))
pretty_print(html("Apakah $a = %s\\in\\mathrm{R17d}?\
\\ %s$"%(latex(a),latex(x in R17d))))
```

$x = 4.369$

$a = \frac{35799}{8192}$

Apakah $x = 4.369 \in \mathbb{Z}?$ False

Apakah $a = \frac{35799}{8192} \in \mathbb{Z}?$ False

Apakah $x = 4.369 \in \mathbb{Q}?$ True

Apakah $a = \frac{35799}{8192} \in \mathbb{Q}?$ True

Apakah $x = 4.369 \in \mathrm{RealField()}?$ True

Apakah $a = \frac{35799}{8192} \in \mathrm{RealField()}?$ True

Apakah $x = 4.369 \in \mathrm{R17d}?$ True

Apakah $a = \frac{35799}{8192} \in \mathrm{R17d}?$ True

Dari hasil pembahasa terlihat bahwa konversi alami dari bilangan rasional dilakukan:

In [10]:
```
a=RDF(8/7)
R100 = RealField(100)
b=R100(8/7)
pretty_print(html("$a =\\mathrm{RDF(8/7)} = %s$"%latex(a)))
pretty_print(html("$b =\\mathrm{R100(8/7)} = %s$"%latex(b)))
```

$a = \mathrm{RDF(8/7)} = 1.1428571428571428$

$b = \mathrm{R100(8/7)} = 1.1428571428571428571428571429$

Seperti konversi antara himpunan bilangan-bilangan *floating-point* yang berbeda:

In [11]:
```
x = R100(8/3)
R = RealField()
pretty_print(html("$\\mathrm{R(x)} = %s$"%latex(R(x))))
pretty_print(html("$\\mathrm{RDF(x)} = %s$"%latex(RDF(x))))
```

$\mathrm{R(x)} = 2.66666666666667$

$\mathrm{RDF(x)} = 2.6666666666666665$

Himpunan bilangan-bilangan *floating-point* yang berbeda berisi nilai-nilai khusus **+0,−0,+infinity,−infinity +0,−0,+infinity,−infinity** dan **NaN**:

In [12]:
```
pretty_print(html("$1.0/0.0=\\mathrm{%s}"%latex(1.0/0.0)))
pretty_print(html("$\\mathrm{RDF(1)/RDF(0)}=\\mathrm{%s}"\
                  %latex(RDF(1)/RDF(0))))
pretty_print(html("$\\mathrm{RDF(-1.)/RDF(0.)}=\\mathrm{%s}"\
                  %latex(RDF(-1.)/RDF(0.))))
```

$1.0/0.0 = +\mathrm{infinity}$

$\mathrm{RDF(1)/RDF(0)} = +\mathrm{infinity}$

$\mathrm{RDF(-1.)/RDF(0.)} = -\mathrm{infinity}$

Nilai khusus **NaN** adalah singkatan dari hasil yang tidak didefinisikan:

In [13]:
```
pretty_print(html("0.0/0.0 = %s"%(0.0/0.0)))
pretty_print(html("RDF(0.0)/RDF(0.0) = %s"%(RDF(0.0)/RDF(0.0))))
```

0.0/0.0 = NaN

RDF(0.0)/RDF(0.0) = NaN

Siktaks : RealField (prec, sci_not, rnd):

Input:

- prec - presisi (integer); default = 53 prec adalah jumlah bit yang digunakan untuk mewakili mantissa dari bilangan floating-point. Presisi dapat berupa bilangan bulat antara mpfr_prec_min() dan mpfr_prec_max(). Dalam implementasi saat ini, mpfr_prec_min() sama dengan 2.
- sci_not - (default: False) jika True, selalu ditampilkan menggunakan notasi ilmiah; jika False, tampilan menggunakan notasi ilmiah hanya untuk bilangan yang sangat besar atau sangat kecil.
- rnd - (string) mode pembulatan
    - 'RNDN' - (default) pembulatan ke terdekat (ikatan menuju ke bilangan genap): Knuth mengatakan ini adalah pilihan terbaik untuk mencegah “floating point drift”.
    - 'RNDD' - pembulatan menuju minus takhingga;
    - 'RNDZ' - pembulat menuju nol.

- 'RNDU' - pembulat menuju plus takhingga ke arah bilangan representatif terdekat: ini adalah apa yang dilakukan dalam himpunan RDF, dan itu adalah perilaku default dari himpunan yang dibuat oleh RealField. Untuk bilangan tepat di tengah dua bilangan yang dapat diwakili, pembulatan dilakukan pada tingkat terdekat bahkan signifikan;

Untuk pembulatan:

Berikut ini contoh-contoh:

In [14]:
```
# Perintah untuk menampilkan himpunan bilangan
# real  RealField() bisa diganti oleh RR
RR==RealField()
```

Out[14]:
```
True
```

In [15]:
```
# Menampilkan himpunan bilangan real dengan presisi 53-bit

RR
```

Out[15]:
```
Real Field with 53 bits of precision
```

In [16]:
```
RealField(100)
```

Out[16]:
```
Real Field with 100 bits of precision
```

In [17]:
```
## Perhatikan bahwa perintah RR dan RealField(100) tidak sama:

RR==RealField(100)
```

Out[17]:
```
False
```

Beberapa Sifat

Pembulatan, yang diperlukan untuk himpunan bilangan-bilngan *floating-point*, menimbulkan banyak efek yang tidak terduga. Kita jelajahi beberapa di antaranya:

Fenomena Berbahaya. Dikenal sebagai bencana kanselasi (*catastrophic cancellation*), ini adalah hilangnya presisi yang dihasilkan dari pengurangan dua bilangan yang sangat dekat; lebih tepatnya, ini merupakan kesalahan yang diperbesar:

In [18]:
```
a = 10000.0
b = 9999.5
c = 0.1
print("a = ",a)
print("b = ",b)
print("c = ",c)
```

```
a =  10000.0000000000
b =  9999.50000000000
c =  0.100000000000000
```

In [19]:
```
a1=a+c                  # penambahan suatu pertubasi kecil ke a
print("a1 = a+c =",a1)
print("a1-b =",a1-b)
```

```
a1 = a+c = 10000.1000000000
a1-b = 0.600000000000364
```

Di sini, kesalahan c yang diperkenalkan pada a membuat perhitungannya tidak tepat (3 digit terakhir salah). Seharusnya : **a+c-b=0.600000000000000**

Coba ketik **a-b+c**, aka memberikan hasil yang diharapkan yaitu **a-b+c=0.600000000000000**.

Hal menjelaskan bahwa secara umum **a+c-b** tidak sama dengan **a-b+c**.

In [20]:
```
print("a-b+c = ",a-b+c)
print("a+c-b =",a+c-b)
```

```
a-b+c =  0.600000000000000
a+c-b = 0.600000000000364
```

Aplikasi: akar-akar Persamaan Kuadrat. Bahkan menghitung akar dari persamaan kuadrat dapat menyebabkan masalah. Kita perhatikan kasus $a = 1.0, b = 10^4, c = 1.0$:

In [21]:
```
a = 1.0; b = 10.0^4; c = 1.0
D = b^2-4*a*c
x_1 = (-b-sqrt(D))/(2*a)
x_2 = (-b+sqrt(D))/(2*a)
print("a = ",a, ",","b = ",b,"dan","c = ",c)
print("D = b^2-4*a*c =",D)
print("x_1 = (-b-sqrt(D))/(2*a) = ",x_1)
print("x_2 = (-b+sqrt(D))/(2*a) =",x_2)
```

```
a =  1.00000000000000 , b =  10000.0000000000 dan c =  1.00000000000000
D = b^2-4*a*c = 9.99999960000000e7
x_1 = (-b-sqrt(D))/(2*a) =  -9999.99990000000
x_2 = (-b+sqrt(D))/(2*a) = -0.000100000001111766
```

Jumlah kedua akar $x_1 + x_2$ benar, tetapi hasil kalinya $x_1 x_2$ tidak benar:

In [22]:
```
print("Apakah x_1+x_2 - (-b/a) = 0 ?",x_1+x_2 - (-b/a) == 0)
print("Apakah x_1*x_2 - c/a = 0 ?",x_1*x_2 - c/a == 0)
print("x_1*x_2-c/a = ",x_1*x_2-c/a)
```

```
Apakah x_1+x_2 - (-b/a) = 0 ? True
Apakah x_1*x_2 - c/a = 0 ? False
x_1*x_2-c/a =  1.11766307320238e-9
```

Kesalahan ini disebabkan oleh fenomena yang dikenal sebagai kanselasi katastropik yang muncul ketika kita menambahkan **-b** dan **sqrt(D)** untuk menghitung **x_2**. Di sini, kita dapat mencoba menemukan perkiraan yang lebih baik untuk **x_2**:

In [23]:
```
x_2 = (c/a)/x_1
print('x_2 =',x_2)
print("x_1+x_2+b/a = ",x_1+x_2+b/a)
print("x_1*x_2-c/a = ",x_1*x_2-c/a)
```

```
x_2 = -0.000100000001000000
x_1+x_2+b/a =  0.000000000000000
x_1*x_2-c/a =  -1.11022302462516e-16
```

In [24]:
```
print("Apakah x_1+x_2 - (-b/a) = 0 ?",x_1+x_2 - (-b/a) == 0)
print("Apakah x_1*x_2 - c/a = 0 ?",x_1*x_2 - c/a == 0)
print("x_1*x_2-c/a = ",x_1*x_2-c/a)
```

```
Apakah x_1+x_2 - (-b/a) = 0 ? True
Apakah x_1*x_2 - c/a = 0 ? False
x_1*x_2-c/a =  -1.11022302462516e-16
```

Disini perlu diperhatikan bahwa, karena pembulatan, jumlah akar-akar tetap benar, tetapi hasil perkaliannya jauh lebih dekat dengan **c/a**. Pembaca dapat mempertimbangkan semua pilihan yang berbeda untuk **a, b** dan **c** harus diyakinkan bahwa menulis program yang kuat secara numerik untuk menghitung akar trinomial kuadrat jauh dari mudah.

Himpunan Bilangan *Floating-Point* bukan Grup Aditif. Sebenarnya, selain itu juga tidak asosiatif. Kita gunakan himpunan **R2** (dengan 2 bit presisi):

In [25]:
```
R2 = RealField(2)
x1 = R2(1/2); x2 = R2(4); x3 = R2(-4)
print("x1 = ",x1)
print("x2 = ",x2)
print("x3 = ",x3)
print("x1+(x2+x3) = ",x1+(x2+x3))
print("(x1+x2)+x3 = ",(x1+x2)+x3)
```

```
x1 =  0.50
x2 =  4.0
x3 =  -4.0
x1+(x2+x3) =  0.50
(x1+x2)+x3 =  0.00
```

Kita dapat menyimpulkan bahwa **berbagai urutan perhitungan dalam suatu program memiliki arti penting pada hasilnya!**

Rekurensi dan Barisan Bilangan *Floating-Point*. Diberikan rekurensi: $u_{n+1} = 4u_n - 1$. Jika $u_0 = 1/3$, urutannya stasioner: $u_i = 1/3$ untuk semua i.

In [26]:
```
x = RDF(1/3)
for i in range(1,100):
    x = 4*x-1; print(x)
```

```
0.33333333333333326
0.33333333333333304
0.33333333333333215
0.3333333333333286
0.3333333333333144
0.33333333333325754
0.33333333333303017
0.3333333333321207
0.3333333333284827
0.3333333333139308
0.3333333332557231
0.3333333330228925
0.3333333320915699
0.3333333283662796
0.3333333134651184
0.33333325386047363
0.33333301544189453
0.3333320617675781
0.3333282470703125
0.33331298828125
0.333251953125
0.3330078125
0.33203125
0.328125
0.3125
0.25
0.0
-1.0
-5.0
-21.0
-85.0
-341.0
-1365.0
-5461.0
-21845.0
-87381.0
-349525.0
-1398101.0
-5592405.0
-22369621.0
-89478485.0
-357913941.0
-1431655765.0
-5726623061.0
-22906492245.0
-91625968981.0
-366503875925.0
-1466015503701.0
-5864062014805.0
-23456248059221.0
-93824992236885.0
-375299968947541.0
```

```
-1501199875790165.0
-6004799503160661.0
-2.4019198012642644e+16
-9.607679205057058e+16
-3.843071682022823e+17
-1.5372286728091292e+18
-6.148914691236517e+18
-2.4595658764946067e+19
-9.838263505978427e+19
-3.935305402391371e+20
-1.5741221609565483e+21
-6.296488643826193e+21
-2.5185954575304773e+22
-1.007438183012191e+23
-4.029752732048764e+23
-1.6119010928195055e+24
-6.447604371278022e+24
-2.5790417485112088e+25
-1.0316166994044835e+26
-4.126466797617934e+26
-1.6505867190471736e+27
-6.602346876188694e+27
-2.6409387504754778e+28
-1.0563755001901911e+29
-4.2255020007607644e+29
-1.6902008003043058e+30
-6.760803201217223e+30
-2.7043212804868892e+31
-1.0817285121947557e+32
-4.326914048779023e+32
-1.730765619511609e+33
-6.923062478046436e+33
-2.7692249912185746e+34
-1.1076899964874298e+35
-4.430759985949719e+35
-1.7723039943798877e+36
-7.089215977519551e+36
-2.8356863910078204e+37
-1.1342745564031281e+38
-4.5370982256125126e+38
-1.814839290245005e+39
-7.25935716098002e+39
-2.903742864392008e+40
-1.1614971457568032e+41
-4.645988583027213e+41
-1.8583954332108852e+42
-7.433581732843541e+42
```

Hasil yang dihitung divergen! Kita dapat mengamati bahwa perilaku ini adalah alami, karena ini adalah fenomena ketidakstabilan klasik: setiap kesalahan pada u_0 dikalikan dengan 4 pada setiap iterasi, dan kita tahu bahwa aritmatika *floating-point* memperkenalkan kesalahan pembulatan, yang akan diperkuat pada setiap iterasi.

Sekarang, kita hitung rekurensi $u_{n+1} = 3u_n - 1$, dengan $u_0 = 1/2$. Diharapkan masalah yang sama: barisan konstan jika dihitung dengan tepat, tetapi setiap kesalahan akan diperkuat pada setiap iterasi.

In [27]:

```
x = RDF(1/2)
for i in range(1,100):
     x = 3*x-1
     print(x)
```

```
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
```

```
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
0.5
```

Sekarang, barisan yang dihitung tetap konstan! Bagaimana kita bisa menjelaskan dua perilaku yang berbeda ini? Kita lihat representasi biner dari u_0 dalam kedua kasus. Untuk kasus pertama ($u_{n+1} = 4u_n - 1, u_0 = 1/3$), maka didapat:

$$\frac{1}{3} = \frac{1}{4}\sum_{i=0}^{\infty}\frac{1}{4^i} = \frac{1}{4}\sum_{i=0}^{\infty}\frac{1}{2^{2i}},$$

dan karenanya $1/3$ tidak dapat direpresentasikan secara tepat dalam himpunan abilangan *floating-point* yang kita miliki. Kita diharap untuk mengulangi perhitungan sebelumnya dalam himpunan dengan presisi besar seperti misalnya **RealField(1000)** untuk memverifikasi bahwa barisan yang dihitung selalu divergen. Selanjutnya jika, dalam program pertama, kita mengganti **x = RDF(1/3)** dengan **x = 1/3**, maka perhitungan dilakukan dalam bilangan rasional dan iterasi tetap sama dengan 1/3. Dalam kasus kedua ($u_{n+1} = 3u_n - 1, u_0 = 1/2$), u_0 dan $3/2$ dalam radix 2 masing-masing 0.1 dan 1.1; karena itu mereka secara tepat diwakili, tanpa pembulatan, dalam himpunan bilangan *floating-point* yang berbeda: perhitungan tepat, dan barisannya tetap konstan.

In [28]:
```
# Berikut ini adalah hasil komputasi eksak

1 - 3*(4/3 - 1)
```

Out[28]: 0

In [29]:
```
# Hasil komputasi numerik: 1 - 3*(4/3. - 1)  tidak sama dengan 0:

1 - (3*(4/3. - 1)) == 0
```

Out[29]: False

In [30]:
```
3*(4/3. - 1)
```

Out[30]: 1.00000000000000

In [31]:
```
N(3*(4/3 - 1))
```

Out[31]: 1.00000000000000

In [32]:
```
1- N(3*(4/3 - 1)) == 0
```

Out[32]: True

In [33]:
```
# Apakah dapat disimpulkan 3*(4/3. - 1) dengan N(3*(4/3 - 1)) sama?

3*(4/3. - 1) == N(3*(4/3 - 1))
```

Out[33]: False

In [34]:
```
R17d = RealField(17,rnd='RNDD')
a = R17d(1)/R17d(3); a.exact_rational()
```

Out[34]: 87381/262144

In [35]:
```
R17u = RealField(17,rnd='RNDU')
a = R17u(1)/R17u(3); a.exact_rational()
```

Out[35]: 43691/131072

In [36]:
```
N(1/3).exact_rational()
```

Out[36]: 6004799503160661/18014398509481984

In [37]:
```
c=RR(1)/RR(3)
c
```

Out[37]: 0.333333333333333

In [38]:
```
1/3 == c.exact_rational()
```

Out[38]: False

In [39]:
```
q = RR(pi/3)
i = cos(q)
arccos(i) == q
```

Out[39]: True

Demikian pula, bilangan desimal tidak tepat direpresentasikan sebagai bilangan biner. Dengan demikian, kita mendapatkan perilaku non-intuitif dalam bentuk program yang seharusnya menghasilkan **satu** sebagaimana beriku ini:

In [40]:
```
a = 0.0
for i in [1..10]:
    a = a + 0.1
a
```

Out[40]: 1.00000000000000

In [41]:
```
# Apakah yakin a=1?

a==1
```

Out[41]: False

Hasilnya adalah logika **False**, hal ini memberitahu kita bahwa program yang telah dibuat tidak sesuai sebagaimana yang kita harapkan. Program yang dibahas adalah menjumlah bilangan desimal **0.1** sebanyak **10 kali**, secara hitungan matematika hasilnya adalah **1**.

Perhatikan bahwa urutan operasi dapat berpengaruh dalam perhitungan sebagaimana berikut ini:

In [42]:
```
b = 1e-16 + 1 - 1e-16
c = 1e-16 - 1e-16 + 1
b == c
```

Out[42]: False

Secara matematika apa yang dibahas ini adalah $b = (1 \times 10^{-16}) + 1 - (1 \times 10^{-16})$ dan $b = (1 \times 10^{-16}) - (1 \times 10^{-16}) + 1$. Dalam hal ini kita tahu bahwa nilai $b = 1$ dan $c = 1$. Tetapi hasil SageMath menyatakan bahwa $b \neq c$. Kira-kira mana salah satu dari b dan c yang mempunyai nilai 1?

In [43]:
```
b,c
```
Out[43]: (1.00000000000000, 1.00000000000000)

In [44]:
```
b==1
```
Out[44]: False

In [45]:
```
c==1
```
Out[45]: True

Terlihat bahwa benar nilai $c = 1$.

Pembaca bisa menyelidiki sendiri, bahwa hasil dari SageMath $b < c$. Diharapkan pembaca bisa memberikan argumentasi mengapa bukan $c < b$. Ada celah antara bilangan floating-point. Ketika bilangannya semakin besar, demikian juga celahnya sebagaimana dibuktikan berikut ini:

In [46]:
```
# Hasil Eksak (benar):

(2^133 + 1) - 2^133
```
Out[46]: 1

In [47]:
```
# Hasil numerik (salah):

(2^133 + 1.0) - 2^133
```
Out[47]: 0.000000000000000

Terlihat bahwa $(2^{133} + 1.0) - 2^{133} = 0$, secara hitungan matematika seharusnya adalah $(2^{133} + 1.0) - 2^{133} = 1$. Para pembaca jangan menyepelehkan masalah ini dengan berargumentasi bahwa nilai bilangan real 1 adalah **sangat-sangat kecil sekali** bila dibandingkan dengan bilangan real 2^{133}. Penulis berpedapat tidak, sebab bila nilai bilangan real 1 kita ganti dengan 10^{12} apakah SageMath bisa tepat menghitung $(2^{133} + 10.0^{12}) - 2^{133}$. Bisa kita uji hal ini sebagai berikut:

In [48]:
```
# Hasil numerik (salah):

(2^133 + 10.0^12) - 2^133
```
Out[48]: 0.000000000000000

In [49]:
```
# Hasil eksak (benar):

(2^133 + 10^12) - 2^133
```

Out[49]:
```
1000000000000
```

Pembahasan semua contoh-contoh bilangan *floating-point* telah dicoba pada **MATLAB R2020a** hasilnya kurang lebih sama. Menurut pendapat saya, **SageMath** lebih unggul dalam komputasi simbolik dan memberikan hasil eksak dalam komputasi. Untuk masalah komputasi numerik **SageMath** lebih lengkap dan rinci dibandingkan dengan **MATLAB R2020a**.

TERIMAKSIH ATAS PERHATIANNYA! SEMOGA BERMANFAAT.

In [50]:
```
e = 1 - 3*(4/3 - 1)
e
```

Out[50]:
```
0
```

[Kembali ke File Utama (http://localhost:8888/notebooks/Workshop_Utama_Unsyiah_2020.ipynb)]

In [ ]:
```

```